code:
@COM92
;乳交
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 92

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;-------------------------------------------------
;射精格子
;-------------------------------------------------
IF CHECK_COM("乳夾口交")
	CALL SAMEN_CHECK, 70
	SELECTCOM = 91
	CALL SAMEN_CHECK, 50
	SELECTCOM = 92
ELSE
	CALL SAMEN_CHECK
ENDIF

RETURN 1

@FLAG_COM92, ARG
SIF CHECK_COM("乳夾口交") == 92
	CALL FLAG_COM91, 50
;対象のＢ⇔調教者のＰ的污垢が移動
CALL EQ_STAIN, TARGET, "Ｂ", PLAYER, "Ｐ"
SIF ASSIPLAY == 0 && EXP:口交経験 >= EXPLV:3
	TFLAG:好感度BOUNS += 1

@SOURCE_COM92, ARG
;乳夾口交
SIF CHECK_COM("乳夾口交") == 92
	CALL SOURCE_COM91, 50

CALL VARSET_COMF

CALL SOURCE_SERVICE_PLAY, "乳交"

CALL SOURCE_DOWNBASE_HP, 92

SOURCE:0:不潔 = COM_ORDER_STAIN("Ｐ", "Ｂ")*40
SOURCE:0:欲情追加 = CALCF("実行値による欲情追加")
SOURCE:0:屈従 = 1800
SOURCE:0:逸脱 = 300

;巨乳
SIF TALENT:巨乳
	SOURCE:0:逸脱 -= 200
;貧乳
SIF TALENT:貧乳
	SOURCE:0:逸脱 += 400

;ABL:Ｂ感覚をみる
SOURCE:0:Ｂ快楽 = CALCF("Ｂ刺激", 2)

SIF TALENT:Ｂ性向
	CALL CALC, "好みのPLAY", SOURCE:0:Ｂ快楽

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM92
IF TALENT:巨乳 && RAND:3 == 0
	LOCALS = %TOFULL(CONDS("罩杯") )%罩杯的%TALENT_NAME(GETNUM(TALENT, "巨乳"), TALENT:巨乳)%
ELSEIF TALENT:男性 == 0 && RAND:2
	LOCALS = 乳房
ELSE
	LOCALS = 胸
ENDIF
IF CHECK_COM("Ｗ奉仕")
	;二人とも前回と同じ行動
	IF SELECTCOM:1 == PREVCOM:1 && SELECTCOM == PREVCOM
		PRINTFORM 兩人
		IF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 && TALENT:淫乱 + TALENT:恋慕 && RAND:3 == 0
			PRINT 神情恍惚、眼神濕潤的對視著的同時
		ELSEIF RAND:2 == 0
			PRINT 如同完全沉浸於快感一樣無意識的漏出涎水的同時
		ENDIF
		PRINTFORMW 將%CALLNAME:MASTER%挺立的陰莖用柔軟的乳肉夾住、持續擼動起來……
	ELSE
		PRINTFORM %CALLNAME:TARGET%和%CALLNAME:(TARGET:1)%
		SIF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 + TALENT:淫乱 + TALENT:恋慕 && RAND:2 == 0
			PRINTFORM 在用充滿慾望與依戀的雙瞳凝視著%CALLNAME:MASTER%赤裸身體的同時、
		IF RAND:3 == 0
			PRINTFORMW 像是在競爭一樣的將陰莖用自己的胸部摩擦著……
		ELSEIF RAND:2 == 0
			PRINTFORMW 用胸部的谷間將陰莖夾住、刺激起來……
		ELSE
			PRINTFORMW 相互協助的將陰莖用胸部擼動起來……
		ENDIF
	ENDIF
ELSEIF CHECK_COM("３Ｐ奉仕")
	CALL TRAIN_MESSAGE_3P
;同時推倒
ELSEIF TEQUIP:Ｗ推倒
	PRINTFORM %CALLNAME:TARGET%
	;助手が前回と同じ行動(違う行動の時は、ここでなく、EQUIP_COM32で記述)
	IF SELECTCOM:1 == PREVCOM:1 && SELECTCOM:1 >= 0
		PRINTFORM 、%TEXTS("Ｗ推倒中")%、
	ENDIF
	IF SELECTCOM == PREVCOM
		PRINTFORMW 將%CALLNAME:MASTER%的陰莖用柔軟的乳肉夾住、持續熱情地摩擦著……
	ELSE
		PRINTFORMW 將%CALLNAME:MASTER%的陰莖用胸部夾住、持續摩擦著……
	ENDIF
;乳夾口交
ELSEIF CHECK_COM("乳夾口交")
	PRINTFORM %CALLNAME:TARGET%
	IF TALENT:恋慕 && RAND:4 == 0
		PRINTFORM 神志恍惚地將%CALLNAME:PLAYER%的
	ELSEIF ABL:奉仕精神 >= 3 && RAND:3 == 0
		PRINTFORM 凝视着%CALLNAME:PLAYER%的脸
	ELSEIF ABL:技巧 >= 3 && RAND:2 == 0
		PRINTFORM 用嫻熟的手法調節節奏將%CALLNAME:PLAYER%
	ELSE
		PRINTFORM %CALLNAME:PLAYER%の
	ENDIF
	IF RAND:4 == 0
		PRINTFORM 龜頭用舌頭反復地舔舐著的同時、用
	ELSEIF RAND:3 == 0
		PRINTFORM 尿道用舌尖小心地刺激著、
	ELSEIF RAND:2
		PRINTFORM 陰莖用舌頭爬上旋轉的舔舐著的同時、用
	ELSE
		PRINTFORM 陰莖用口腔含住、緊緊吮吸著的同時、用
	ENDIF
	;妊娠
	IF TALENT:妊娠 && TALENT:貧乳 == 0 && RAND:3 == 0
		PRINT 因充滿母乳而鼓脹起來的
	;巨乳
	ELSEIF TALENT:巨乳
		;爆乳
		IF TALENT:巨乳 >= 2 && RAND:3 == 0
			PRINT 擁有沉甸甸重量感的
		ELSEIF RAND:2
			PRINT 富含彈性的
		ELSE
			PRINT 丰腴的
		ENDIF
	;貧乳
	ELSEIF TALENT:貧乳
		PRINT 小小的
		LOCALS = 山包
	ELSEIF TALENT:男性 == 0
		IF RAND:3 == 0
			PRINT 具有恰當的彈性的
		ELSEIF RAND:2
			PRINT 有张力
		ELSE
			PRINT 形状好
		ENDIF
	ENDIF

	PRINTFORM %LOCALS%將陰莖
	IF TALENT:巨乳 >= 2 && RAND:2
		PRINTFORM 包裹住
	ELSEIF TALENT:貧乳 == 0 && TALENT:男性 == 0
		PRINTFORM 夾住
	ENDIF
	PRINTFORMW 持續摩擦起來……
;前回と同じ
ELSEIF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
	PRINTFORM %CALLNAME:TARGET%、
	IF TALENT:恋慕 && RAND:4 == 0
		PRINTFORM 神情恍惚的緊緊凝視著%CALLNAME:PLAYER%的同時、將挺立的陰莖
	ELSEIF ABL:奉仕精神 >= 3 && RAND:3 == 0
		PRINTFORM 密切關注著%CALLNAME:PLAYER%神情的同時、將挺立的陰莖
	ELSEIF ABL:技巧 >= 3 && RAND:2
		PRINTFORM 用嫻熟的手法調節節奏的同時、將%CALLNAME:PLAYER%挺立的陰莖
	ELSEIF ABL:精液中毒 >= 3 && RAND:2
		PRINTFORM 用飢渴而狂熱地熱烈動作將%CALLNAME:PLAYER%挺立的陰莖
	ELSE
		PRINTFORM 將%CALLNAME:PLAYER%挺立的陰莖
	ENDIF
	IF TALENT:貧乳 || TALENT:男性
		PRINTFORMW 用%LOCALS%貼緊並持續摩擦著……
	ELSE
		PRINTFORMW 用%LOCALS%持續擼動著……
	ENDIF
;前回は雙乳間抽插
ELSEIF PREVCOM == 81 && COND("前Ｔ調教者") == PLAYER
	PRINTFORM 在%CALLNAME:PLAYER%停止腰肢的挺動、發出用胸部侍奉的要求之後、%CALLNAME:TARGET%
	;恋慕かつ奉仕精神5or献身的かつ奉仕精神3以上
	IF ( (TALENT:恋慕 && ABL:奉仕精神 == 5) || (TALENT:献身 && ABL:奉仕精神 >= 3) ) && RAND:3 == 0
		IF RAND:2 == 0
			PRINT 滿臉幸福快樂的
		ELSE
			PRINT 用不住上翻的眼珠仰視著神情的同時、
		ENDIF
	;恋慕or献身的or奉仕精神3以上
	ELSEIF TALENT:恋慕 || TALENT:献身 || ABL:奉仕精神 >= 3
		PRINT 帶著難以掩藏的開心表情、
	;強気
	ELSEIF TALENT:強気 && RAND:3
		PRINTFORM 用與平日裡強硬眼神截然不同的濕潤瞳孔凝視著%CALLNAME:PLAYER%的同時、
	ENDIF
	IF TALENT:貧乳 || TALENT:男性
		PRINTFORM 用貧瘠單薄的胸膛拼命地將陰莖
	ELSEIF TALENT:巨乳
		PRINTFORM 用豐腴飽滿的果實將挺立的陰莖包裹住、
	ELSE
		PRINTFORM 用%LOCALS%將挺立的陰莖裹挾著、
	ENDIF
	PRINTW 持續摩擦起來……

;气味弱点分岐
ELSEIF PLAYER == MASTER && TALENT:气味弱点 && TALENT:恋慕 && (CHOSECOM(SELECTCOM) == 0 || RAND:3 == 0)
	PRINTFORM %CALLNAME:TARGET%
	IF CHECK_STAIN(STAIN:PLAYER:Ｐ, "Ｐ恥垢") && RANDIF(2, CHECK_STAIN(STAIN:PLAYER:Ｐ, "精液") )
		IF RAND:2
			PRINTFORM 將%CALLNAME:PLAYER%挺立的陰莖
			IF TALENT:貧乳 || TALENT:男性
				PRINTFORMW 上黏著的恥垢所散發著的淫靡%TEXTS("气味")%興奮地深深嗅聞著的同時、用自己的胸部摩擦著刺激起來……
			ELSE
				PRINTFORMW 用自己的胸部夾住、在將濃郁恥垢所散發的淫靡%TEXTS("气味")%投入地深深嗅聞著的同時、用柔軟的乳肉擼動起來……
			ENDIF
		ELSE
			PRINTFORM 在將%CALLNAME:PLAYER%蒸騰著熱汽的股間所散發的%TEXTS("气味")%深深嗅聞著的同時、
			IF TALENT:貧乳 || TALENT:男性
				PRINTFORMW 將挺立的陰莖與自己貧瘠的胸膛緊貼、拼命地摩擦著侍奉起來……
			ELSE
				PRINTFORMW 將挺立的陰莖用自己柔軟的乳肉夾住、投入地擼動著侍奉起來……
			ENDIF
		ENDIF
	ELSEIF CHECK_STAIN(STAIN:PLAYER:Ｐ, "精液") && RAND:2
		IF RAND:2
			PRINTFORM 將被精液弄髒的%CALLNAME:PLAYER%的陰莖
			IF TALENT:貧乳 || TALENT:男性
				PRINTFORMW 與身體緊貼、在將%TEXTS("气味")%氣味細嗅著漸漸興奮起來的同時、用胸部不斷摩擦著侍奉起來……
			ELSE
				PRINTFORMW 用胸部夾住、在將%TEXTS("气味")%氣味細嗅著漸漸興奮起來的同時、用胸部不斷摩擦著侍奉起來……
			ENDIF
		ELSE
			PRINTFORM 將%CALLNAME:PLAYER%那散發著精液%TEXTS("气味")%的髒污陰莖
			IF TALENT:貧乳 || TALENT:男性
				PRINTFORMW 與身體緊貼、帶著快樂而幸福的表情持續用胸部摩擦著侍奉起來……
			ELSE
				PRINTFORMW 用胸部夾住、帶著快樂而幸福的表情將胸部上下搖動著擼動起來……
			ENDIF
		ENDIF
	ELSEIF TALENT:貧乳 || TALENT:男性
		PRINTFORMW 將散發著濃郁的雄性氣味的陰莖用胸部持續侍奉著的同時、將蒸騰著熱汽的%CALLNAME:PLAYER%的股間所散發著的淫靡%TEXTS("气味")%深深嗅聞著並因此而興奮著……
	ELSE
		IF RAND:2
			PRINTFORMW 將散發著濃郁的雄性氣味的陰莖用胸部夾住並持續摩擦著的同時、將蒸騰著熱汽的%CALLNAME:PLAYER%的股間所散發著的淫靡%TEXTS("气味")%深深嗅聞著並因此而興奮著……
		ELSE
			PRINTFORMW 將%CALLNAME:PLAYER%挺立的陰莖用柔軟的乳肉擼動著的同時、也將不時抵近鼻尖的紫紅龜頭所散發著淫靡%TEXTS("气味")%深深嗅聞著……
		ENDIF
	ENDIF
;被逆推ているとき
ELSEIF TEQUIP:推倒
	IF TEQUIP:面部騎乗
		PRINTFORMW %CALLNAME:TARGET%跨坐在%CALLNAME:PLAYER%的臉頰上、將胸口移向股間、
	ELSE
		PRINTFORMW %CALLNAME:TARGET%將%CALLNAME:PLAYER%的下半身用雙手壓住、將胸口移向股間、
	ENDIF
	;淫乱or施虐狂or施虐属性Lv.5以上で「害羞」を持たない
	SIF (TALENT:淫乱 || TALENT:小悪魔 || ABL:施虐属性 >= 5) && TALENT:害羞 == 0
		PRINT 將滴落的晶瑩涎水塗抹上去充當潤滑劑的同時、
	PRINTFORM %CALLNAME:TARGET%
	;胸元が露出しているなら、そこではさむ
	IF CHECK_CLO("谷間露出") && RAND:3
		IF TEQUIP:六九式
			PRINTFORM %TEXTS("６９")%
		ELSEIF TALENT:貧乳
		ELSEIF TALENT:巨乳 && RAND:2 == 0
			PRINT 任由深邃的谷間被一覽無遺、
		ELSE
			PRINT 將胸前的風景大膽地暴露著、
		ENDIF
	;半脱られるのに、胸元を半脱ていないなら、ずらす
	ELSEIF CHECK_CLO("半脱可")
		IF CHECK_SHIRT("半脱可")
			PRINTFORM 將%SHIRT(TARGET, "略")%將捲起、
		ELSE
			PRINT 將胸前的衣襟敞開、
		ENDIF
		CALL SETFLAG, "胸を半脱る", TARGET
	ENDIF
	IF TALENT:貧乳 || TALENT:男性
		PRINT 用貧瘠單薄的胸膛將陰莖
	ELSEIF GETBITS(CFLAG:妊娠で巨乳化, 0, 1) && RAND:2 == 0
		PRINT 用與過去相比膨脹許多的胸部夾住陰莖、
	ELSEIF TALENT:巨乳 && RAND:2
		PRINT 用丰腴柔軟的乳肉將陰莖包裹住、
	ELSE
		PRINTFORM 用%LOCALS%將挺立的陰莖裹挾著、
	ENDIF
	PRINTW 擦好了……
;お浴室(胸は当然剥き出しと想定)
ELSEIF TEQUIP:浴室PLAY
	PRINTFORM %CALLNAME:TARGET%
	IF TALENT:貧乳 || TALENT:男性
		PRINTW 用塗滿了泡沫的單薄胸膛、拼命地將陰莖擦洗著……
	ELSEIF TALENT:巨乳
		PRINTW 用谷間充滿了泡沫的豐腴胸部、將陰莖直達根部的包住、揉搓著擦洗起來……
	ELSE
		PRINTW 將塗滿泡沫的綿軟胸部當做海綿、將陰莖夾住擦洗著……
	ENDIF
;それ以外
ELSE
	PRINTFORM %CALLNAME:TARGET%
	;胸元が露出しているなら、そこではさむ
	IF CHECK_CLO("谷間露出")
		IF TEQUIP:六九式
			PRINTFORM %TEXTS("６９")%
		ELSEIF TALENT:貧乳
		ELSEIF TALENT:巨乳 && RAND:2 == 0
			PRINT 任由深邃的谷間被一覽無遺、
		ELSE
			PRINT 將胸前的風景大膽地暴露著、
		ENDIF
	;半脱られるのに、胸元を半脱ていないなら、ずらす
	ELSEIF CHECK_CLO("半脱可")
		IF CHECK_SHIRT("半脱可")
			PRINTFORM 將%SHIRT(TARGET, "略")%將捲起、
		ELSE
			PRINT 將胸前的衣襟敞開、
		ENDIF
		CALL SETFLAG, "胸を半脱る", TARGET
	ENDIF
	IF TALENT:貧乳 || TALENT:男性
		PRINTFORM 用貧瘠單薄的胸膛拼命地將陰莖
	ELSEIF TALENT:巨乳
		PRINTFORM 用豐腴飽滿的果實將挺立的陰莖包裹住、
	ELSE
		PRINTFORM 用%LOCALS%將挺立的陰莖裹挾著、
	ENDIF
	PRINTW 擦好了……
ENDIF
;施虐属性LV3以上で尿道刺激、口球使用中は不可
SIF ABL:施虐属性 >= 3 && TEQUIP:口枷 == 0 && RAND:2 == 0
	PRINTFORMW 同時、將微微張合著的尿道口用舌尖小心而溫柔地刺激著……

@COM_ORDER_COM92, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF
;ABL:精液中毒
IF ABL:精液中毒
	LOCAL += MIN(ABL:精液中毒, 5) * 3
	LOCALS = %LOCALS% + 精液中毒LV{ABL:精液中毒}({MIN(ABL:精液中毒, 5) * 3})
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 3
	LOCALS = %LOCALS% - 害羞(3)
ENDIF
;献身的
IF TALENT:献身
	LOCAL += 6
	LOCALS = %LOCALS% + 献身(6)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF
;Ｂ性向
IF TALENT:Ｂ性向
	LOCAL += 10
	LOCALS = %LOCALS% + Ｂ性向(10)
ENDIF
;男性
IF TALENT:男性 && TALENT:兩面通吃 == 0
	LOCAL -= 15
	LOCALS = %LOCALS% - 男性(15)
ENDIF

;汚れ有
IF COM_ORDER_STAIN("Ｐ", "Ｂ")
	LOCAL -= COM_ORDER_STAIN("Ｐ", "Ｂ")
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({COM_ORDER_STAIN("Ｐ", "Ｂ")})
ENDIF

;合計を表示(30以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 30

RETURN RESULT


code:
@COM93
;素股
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 93

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%
;-------------------------------------------------
;射精格子
;-------------------------------------------------
CALL SAMEN_CHECK

RETURN 1

@FLAG_COM93, ARG
IF CHECK_COM("展示肉棒")
	CALL SETFLAG, "展示肉棒フラグ", TARGET
ELSE
	;対象のＶ⇔調教者のＰ的污垢が移動
	SIF CHECK_CLO("Ｖ露出")
		CALL EQ_STAIN, TARGET, "Ｖ", PLAYER, "Ｐ"
ENDIF
SIF ASSIPLAY == 0
	TFLAG:好感度BOUNS += 1

@SOURCE_COM93, ARG
CALL VARSET_COMF

SOURCE:0:欲情追加 = CALCF("実行値による欲情追加")
SOURCE:0:屈従 = 1200
SOURCE:0:逸脱 = 50

IF CHECK_COM("展示肉棒") == 0
	CALL SOURCE_SERVICE_PLAY, "素股"

	CALL SOURCE_DOWNBASE_HP, 93

	SOURCE:0:不潔 = COM_ORDER_STAIN("Ｐ", "Ｖ")*50 + 60

	;ABL:Ｃ感覚をみる
	SOURCE:0:Ｃ快楽 = CALCF("Ｃ刺激", 2)
	;阴茎への刺激にはBOUNS
	SIF PENIS(TARGET)
		SOURCE:0:Ｃ快楽 += 500

	;PALAM:Ｖ潤
	SELECTCASE PALAM:Ｖ潤
	CASE IS < PALAMLV:1
		TIMES SOURCE:0:Ｃ快楽, 0.10
		TIMES SOURCE:0:性行動, 0.60
	CASE IS < PALAMLV:2
		TIMES SOURCE:0:Ｃ快楽, 0.30
		TIMES SOURCE:0:性行動, 0.80
	CASE IS < PALAMLV:3
		TIMES SOURCE:0:Ｃ快楽, 0.50
		TIMES SOURCE:0:性行動, 1.00
	CASE IS < PALAMLV:4
		TIMES SOURCE:0:性行動, 1.20
	CASE IS < PALAMLV:5
		TIMES SOURCE:0:Ｃ快楽, 0.90
		TIMES SOURCE:0:性行動, 1.40
	CASEELSE
		TIMES SOURCE:0:Ｃ快楽, 0.80
		TIMES SOURCE:0:性行動, 1.60
	ENDSELECT

	;阴部が露出していないと快Ｃ減
	IF CHECK_CLO("Ｖ露出") == 0 && COND("調教中")
		;休闲褲等を是ていると快Ｃ激減
		IF CHECK_OUTER("休闲褲") || CHECK_PANTIES("皮革")
			TIMES SOURCE:0:Ｃ快楽, 0.20
			TIMES SOURCE:0:露出, 0.20
		ELSE
			TIMES SOURCE:0:Ｃ快楽, 0.60
			TIMES SOURCE:0:露出, 0.50
		ENDIF
	ENDIF
ENDIF

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM93
;ちんこの尺寸は２以上
IF CHECK_COM("展示肉棒")
	PRINTFORMW %CALLNAME:PLAYER%挺起腰肢将%TEXTS("巨根：視覚", PLAYER)%向%CALLNAME:TARGET%展示着。
	IF ABL:欲望 < 3 || EXP:Ｖ性交経験 + EXP:Ａ性交経験 + EXP:尿道性交経験 == 0
		PRINTFORMW %CALLNAME:TARGET%在%CALLNAME:PLAYER%將陰莖壓上的時候、眼神一直沒能離開……
	ELSE
		PRINTFORM %CALLNAME:TARGET%將%CALLNAME:PLAYER%的陰莖%TEXTR("用腰肢輕輕摩擦著/放在腹部感受著")%、
		PRINTFORMW %TEXTR("因這雄偉而神往著/臉紅地想像著接下來的/確認著其中的灼熱/忍耐著愈加旺盛的慾望")%……
	ENDIF
ELSEIF CHECK_COM("Ｗ奉仕")
	IF SELECTCOM:1 == PREVCOM:1 && SELECTCOM == PREVCOM
		PRINTFORM 兩人
		IF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 && TALENT:淫乱 + TALENT:恋慕 && RAND:3 == 0
			PRINT 神情恍惚、眼神濕潤的對視著的同時
		ELSEIF RAND:2 == 0
			PRINT 如同完全沉浸於快感一樣無意識的漏出涎水的同時
		ENDIF
		PRINTFORMW 將%CALLNAME:MASTER%的陰莖用股間貼緊摩擦著、持續地給予並索取著快樂……
	ELSE
		PRINTFORM %CALLNAME:TARGET%和%CALLNAME:(TARGET:1)%
		SIF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 + TALENT:淫乱 + TALENT:恋慕 && RAND:2 == 0
			PRINTFORM 在用充滿慾望與依戀的雙瞳凝視著%CALLNAME:MASTER%赤裸身體的同時、
		PRINTFORMW 互相协力用\@ PENIS(TARGET) + PENIS(TARGET:1) == 0 ? 阴部 # 股間 \@将阴茎夹紧…
	ENDIF
ELSE
	IF TEQUIP:接吻
		PRINTFORM %CALLNAME:TARGET%將%CALLNAME:PLAYER%%TEXTS("接吻継続助詞付き")%將阴茎用
		CALL SETFLAG, "追加接吻フラグ", TARGET
	ELSE
		PRINTFORM %CALLNAME:TARGET%將阴茎用
	ENDIF
	IF PREVCOM == SELECTCOM && PREVPLAY("展示肉棒") == 0
		PRINTFORMW 柔軟的股間夾住、持續摩擦著……
	ELSE
		;阴部剥き出し
		IF CHECK_CLO("Ｖ露出") && TALENT:男性 == 0
			IF RAND:2
				PRINT 豐腴而柔軟的
			ELSEIF COND("陰毛") >= 6
				PRINT 濡濕的陰毛糾纏在一起的
			ELSEIF COND("陰毛") == 5
				;PRINT 綺麗に剃り整えられた
				PRINT 十分柔軟的
			;ELSEIF COND("陰毛") == 4
			;	PRINT 还没長出来の
			ELSEIF COND("陰毛") && COND("陰毛") <= 3
				PRINT 剝殼雞蛋一樣光潔的
			ENDIF
		ENDIF
		PRINTFORM 胯股之间夾住、
		SIF ABL:腰靈活 >= 3
			PRINT 腰肢淫靡的擺動著
		IF CHOSECOM(SELECTCOM) == 0 && PENIS(PLAYER) && SIZE("阴茎", PLAYER) >= 2
			PRINTFORMW 開始摩擦。
			PRINTFORM %CALLNAME:TARGET%在侍奉的間歇將%CALLNAME:PLAYER%的陰莖用自己的腹部貼緊、
			IF CHOSECOMS("挿入")
				PRINTFORMW 丈量著體內進入了多少……
			ELSE
				PRINTFORMW 丈量著會抵達哪裡……
			ENDIF
		ELSE
			PRINTFORMW 摩擦起來…
		ENDIF
	ENDIF
ENDIF

@COM_ORDER_COM93, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF
;ABL:精液中毒
IF ABL:精液中毒
	LOCAL += MIN(ABL:精液中毒, 5) * 3
	LOCALS = %LOCALS% + 精液中毒LV{ABL:精液中毒}({MIN(ABL:精液中毒, 5) * 3})
ENDIF

;害羞
IF TALENT:害羞
	LOCAL -= 3
	LOCALS = %LOCALS% - 害羞(3)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF

;媚薬
IF TEQUIP:媚薬
	LOCAL += 6
	LOCALS = %LOCALS% + 媚薬(6)
ENDIF

;汚れ有
IF COM_ORDER_STAIN("Ｐ", "Ｖ")/3
	LOCAL -= COM_ORDER_STAIN("Ｐ", "Ｖ")/3
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({COM_ORDER_STAIN("Ｐ", "Ｖ")/3})
ENDIF

;合計を表示(20以上で実行)
IF CHECK_COM("展示肉棒", 93)
	CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 5
ELSE
	CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 20
ENDIF

RETURN RESULT

license: unknown


code:
@COM94
;泡泡浴
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 94

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;-------------------------------------------------
;射精格子
;-------------------------------------------------
CALL SAMEN_CHECK

RETURN 1

@FLAG_COM94, ARG
;汚れを初期化
CALL RESET_STAIN_EX, "全部", PLAYER
CALL RESET_STAIN_EX, "全部", TARGET

@SOURCE_COM94, ARG
CALL VARSET_COMF

CALL SOURCE_SERVICE_PLAY, "泡泡浴"

CALL SOURCE_DOWNBASE_HP, 94

SOURCE:0:欲情追加 = CALCF("実行値による欲情追加")
SOURCE:0:屈従 = 2000
SOURCE:0:逸脱 = 1500

;ABL:Ｃ感覚をみる
SOURCE:0:Ｃ快楽 = CALCF("Ｃ刺激", 2, 50)
;ABL:Ｂ感覚をみる
SOURCE:0:Ｂ快楽 = CALCF("Ｂ刺激", 2, 50)

SIF TALENT:巨乳
	TIMES SOURCE:0:Ｂ快楽, 1.20

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM94
IF CHECK_COM("Ｗ奉仕")
	;二人とも前回と同じ行動
	IF SELECTCOM:1 == PREVCOM:1 && SELECTCOM == PREVCOM
		PRINTFORM 兩人
		IF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 && TALENT:淫乱 + TALENT:恋慕 && RAND:3 == 0
			PRINT 神情恍惚、眼神濕潤的對視著的同時
		ELSEIF RAND:2 == 0
			PRINT 如同完全沉浸於快感一樣無意識的漏出涎水的同時
		ENDIF
		PRINTFORMW 將%CALLNAME:MASTER%用沾滿泡沫的胸部持續擦洗著身體……
	ELSE
		PRINTFORM %CALLNAME:TARGET%和%CALLNAME:(TARGET:1)%
		SIF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 + TALENT:淫乱 + TALENT:恋慕 && RAND:2 == 0
			PRINTFORM 在用充滿慾望與依戀的雙瞳凝視著%CALLNAME:MASTER%赤裸身體的同時、
		IF RAND:2
			PRINTFORM 像是在競爭一樣的將泡沫塗滿自己的
		ELSE
			PRINTFORM 相互協助的將泡沫塗滿自己的
		ENDIF
		;それぞれの身体の差も描写すべきかなー
		IF RAND:3 == 0
			IF RAND:3 == 0
				PRINT 胸部
			ELSEIF RAND:2
				PRINT 乳房
			ELSE
				PRINT 胸
			ENDIF
		ELSEIF RAND:2
			PRINT 将身体
		ELSE
			PRINT 股間
		ENDIF
		PRINTW 來協助擦洗著…
	ENDIF
;同時推倒
ELSEIF TEQUIP:Ｗ推倒
	PRINTFORM %CALLNAME:TARGET%
	;助手が前回と同じ行動(違う行動の時は、ここでなく、EQUIP_COM32で記述)
	IF SELECTCOM:1 == PREVCOM:1 && SELECTCOM:1 >= 0
		PRINTFORM 、%TEXTS("Ｗ推倒中")%、
	ENDIF
	IF SELECTCOM == PREVCOM
		PRINTFORMW 將%CALLNAME:MASTER%的身體抱住、用自己塗滿泡沫的身體持續熱情的摩擦著……
	ELSE
		PRINTFORMW 將%CALLNAME:MASTER%的身體用塗滿泡沫的胸部擦洗起來……
	ENDIF
;前回と同じ
ELSEIF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
	PRINTFORM %CALLNAME:TARGET%
	IF TEQUIP:接吻
		PRINTFORM 在%TEXTS("接吻継続")%、
		CALL SETFLAG, "追加接吻フラグ", TARGET
	ENDIF
	IF TALENT:貧乳 == 0 && RAND:3 == 0
		PRINTFORMW 將塗滿泡沫的綿軟胸部當做海綿、將%CALLNAME:PLAYER%的手腕和臉頰夾住擦洗起來……
	ELSEIF RAND:2
		PRINTFORMW 用沾滿泡沫的股間將手腕夾住、一邊享受著%CALLNAME:PLAYER%回禮一樣的愛撫一邊協助擦洗著身體
	ELSE
		PRINTFORMW 為%CALLNAME:PLAYER%的身體塗抹上泡沫、\@ ABL:奉仕精神 >= 3 ? 熱情地 # \@持續擦洗著……
	ENDIF
ELSE
	PRINTFORM %CALLNAME:TARGET%
	IF RAND:2 == 0
		IF RAND:2 == 0
			PRINTFORM 跨坐在%CALLNAME:PLAYER%的手腕上、用自己沾滿泡沫的
		ELSE
			PRINTFORM 跨坐在仰躺在塑膠墊上的%CALLNAME:PLAYER%的背上、用自己沾滿泡沫的
		ENDIF
		SELECTCASE COND("陰毛")
		CASE IS >= 6
			PRINT 陰毛
		CASE 1 TO 3
			PRINT 光潔無毛的股間
		CASEELSE
			PRINT 股間
		ENDSELECT
		PRINTW 擦洗起來……
	ELSE
		PRINTFORM 將能打起泡沫的肥皂胡亂地塗抹在身上、
		SIF TALENT:恋慕
			PRINT 在陶醉地凝視著的同時

		IF TALENT:妊娠 && RAND:2
			PRINTFORM 用因%CALLNAME:PLAYER%而懷孕膨脹起來的腹部
		ELSEIF TALENT:巨乳 && RAND:2
			IF TALENT:娇小
				PRINTFORM 用和身體相比過分豐滿的胸部
			ELSE
				PRINTFORM 用豐腴柔軟的胸部
			ENDIF
		ELSEIF TALENT:娇小 && RAND:2
			PRINTFORM 用嬌小的身體
		ELSEIF RAND:2
			PRINTFORM 用全身
		ELSE
			PRINTFORM 将身体
		ENDIF
		SIF ABL:奉仕精神 >= 3
			PRINTFORM 全心地
		PRINTW 來協助擦洗著…
	ENDIF
ENDIF

@COM_ORDER_COM94, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 3
	LOCALS = %LOCALS% - 害羞(3)
ENDIF
;献身的
IF TALENT:献身
	LOCAL += 6
	LOCALS = %LOCALS% + 献身(6)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF

;汚れ有
IF COM_ORDER_STAIN("Ｐ", "Ｖ")
	LOCAL -= COM_ORDER_STAIN("Ｐ", "Ｖ")
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({COM_ORDER_STAIN("Ｐ", "Ｖ")})
ENDIF

;合計を表示(30以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 30

RETURN RESULT


code:
@COM95
;足交
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 95

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;-------------------------------------------------
;射精格子
;-------------------------------------------------
CALL SAMEN_CHECK

RETURN 1

@FLAG_COM95, ARG

@SOURCE_COM95, ARG
CALL VARSET_COMF

CALL SOURCE_SERVICE_PLAY, "足交"

CALL SOURCE_DOWNBASE_HP, 95

IF PENIS(PLAYER)
	SOURCE:0:不潔 = COM_ORDER_STAIN("Ｐ", "脚")*10 + 60
ELSE
	SOURCE:0:不潔 = COM_ORDER_STAIN("Ｖ", "脚")*10 + 60
ENDIF
SOURCE:0:欲情追加 = CALCF("実行値による欲情追加")
SOURCE:0:屈従 = 400
SOURCE:0:逸脱 = 400

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM95
;TEQUIP:足交のbitで行動管理
;0bit 踏みつけ,1bit 股間見せ, 2bit 両足
IF CHECK_COM("Ｗ奉仕")
	;二人とも前回と同じ行動
	IF SELECTCOM:1 == PREVCOM:1 && SELECTCOM == PREVCOM
		PRINTFORM 兩人
		IF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 && TALENT:淫乱 + TALENT:恋慕 && RAND:3 == 0
			PRINT 神情恍惚、眼神濕潤的對視著的同時
		ELSEIF RAND:2 == 0
			PRINT 如同完全沉浸於快感一樣無意識的漏出涎水的同時
		ENDIF
		PRINTFORM %CALLNAME:MASTER%的
		IF PENIS(PLAYER)
			PRINT 阴茎
		ELSE
			PRINTFORM \@ PREVPLAY("Ｖ挿入", MASTER) ? 黏黏糊糊被放鬆的 # \@阴部
		ENDIF
		PRINTFORMW 用兩隻靈巧柔軟的纖足蹂躪玩弄著……
	ELSE
		PRINTFORM %CALLNAME:TARGET%和%CALLNAME:(TARGET:1)%
		IF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 && TALENT:淫乱 + TALENT:恋慕
			IF RAND:3 == 0
				PRINTFORM 在用充滿慾望與依戀的雙瞳凝視著%CALLNAME:MASTER%赤裸身體的同時、
			ELSEIF RAND:2 == 0
				PRINTFORM 保持著能將股間清楚看見的良好姿勢、
				SETBIT TEQUIP:足交, 1
			ENDIF
		ENDIF
		SIF RAND:2 == 0
			PRINT 相互協助的
		IF PENIS(PLAYER)
			PRINTFORMW 將陰莖用兩隻靈活的纖足擼動起來……
		ELSE
			PRINTFORMW 將陰部用兩隻靈活的纖足戲弄起來……
		ENDIF
	ENDIF
;同時推倒
ELSEIF TEQUIP:Ｗ推倒
	PRINTFORM %CALLNAME:TARGET%
	;助手が前回と同じ行動(違う行動の時は、ここでなく、EQUIP_COM32で記述)
	SIF SELECTCOM:1 == PREVCOM:1 && SELECTCOM:1 >= 0
		PRINTFORM %TEXTS("Ｗ推倒中")%、
	PRINTFORM %CALLNAME:MASTER%的
	IF PENIS(PLAYER)
		PRINT 陰莖
	ELSE
		PRINT 陰部
	ENDIF
	IF ABL:施虐属性 >= 3 && ABL:技巧 >= 3 && RAND:2 == 0
		PRINT 技巧嫻熟的
	ELSEIF ABL:施虐属性 >= 3
		PRINT 像是在享受痛苦的反應一樣粗暴地
	ELSEIF ABL:技巧 >= 3 && PENIS(PLAYER)
		PRINT 扑哧扑哧地
	ELSEIF ABL:施虐属性 < 3 || ABL:技巧 < 3
		PRINT 笨拙而小心的
	ENDIF
	IF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
		IF PENIS(PLAYER)
			PRINTW 持續擼動著……
		ELSE
			PRINTW 持續蹂躪著……
		ENDIF
	ELSE
		IF PENIS(PLAYER)
			PRINTW 擼動著……
		ELSE
			PRINTW 蹂躪著……
		ENDIF
	ENDIF
ELSE
	PRINTFORM %CALLNAME:TARGET%
	;被逆推中
	IF TEQUIP:推倒
		SIF PENIS(PLAYER) == 0
			SETBIT TEQUIP:足交, 0
		IF TEQUIP:面部騎乗 && ABL:施虐属性 >= 3
			PRINTFORM 保持著將%CALLNAME:PLAYER%的臉頰用股間壓迫著的姿勢、
			TEQUIP:足交 = 0
		ELSEIF SELECTCOM == PREVCOM
		;施虐属性3以上
		ELSEIF ABL:施虐属性 >= 3 && RAND:2 && GETBIT(TEQUIP:足交, 0) == 0
			PRINTFORM %CALLNAME:PLAYER%の
			IF TALENT:娇小
				PRINT 胸
			ELSE
				PRINT 神情恍惚的脸颊
			ENDIF
			PRINT 將腰肢毫不留情地重重落下、向下半身伸出了雙腳、
		ELSEIF GETBIT(TEQUIP:足交, 0)
			IF ABL:施虐属性 >= 3 && COND("脚の使用", PLAYER) && RAND:2
				PRINTFORM 將%CALLNAME:PLAYER%放倒在地、將雙腿腳踝抓緊後高高舉起、做出準備進行電氣按摩的姿勢、
			ELSE
				PRINTFORM 將%CALLNAME:PLAYER%佈滿緊張地臉頰
				SIF ABL:施虐属性 >= 3
					PRINT 愉快的邪笑着
				PRINT 俯瞰的同時、
			ENDIF
			;面部騎乗解除
			TEQUIP:面部騎乗 = 0
		ENDIF
	;休闲褲
	ELSEIF EQUIP:外套 && CHECK_OUTER("休闲褲")
	;露出癖が低め
	ELSEIF ABL:露出癖 - TALENT:害羞 + TALENT:不知羞恥 + TALENT:好色 + TALENT:淫乱 + TALENT:小悪魔 < RAND:4
		PRINT 若無其事地將股間用
		;圍裙
		IF TEQUIP:新妻PLAY
			PRINT 圍裙的下擺
		;裙のある服
		ELSEIF EQUIP:裙
			PRINTFORM %SKIRT(TARGET, "略")%
		ELSE
			PRINTFORM 雙手
		ENDIF
		PRINT 遮掩著的同時、
	;露出癖3以上or施虐狂or淫乱で休闲褲(これは上のほうで弾き済み)でない
	ELSEIF ABL:奉仕精神 + TALENT:献身*2 >= 4 || ABL:露出癖 + ABL:施虐属性 >= 3 || TALENT:好色 + TALENT:淫乱 + TALENT:恋慕 + TALENT:小悪魔 || PALAM:欲情 >= 10000
		;裙などを是ているなら
		IF EQUIP:裙 + TEQUIP:新妻PLAY && RAND:3
			;圍裙
			IF TEQUIP:新妻PLAY
				PRINT 將圍裙
			ELSE
				PRINTFORM 將%SKIRT(TARGET, "略")%
			ENDIF
			IF COND("口の使用") && RANDIF(2, COND("指の使用") )
				PRINT 捲起、將下擺用嘴含住、將
			ELSEIF COND("指の使用") && RAND:2
				PRINT 用指尖捏起下擺、將
			ELSE
				PRINT 卷起、
			ENDIF
		;胖次を是ているなら、ずらすことも
		ELSEIF !CHECK_CLO("Ｖ露出") && CHECK_CLO("Ｖ露出可") && RAND:5 == 0
			CALL SETFLAG, "ボトムを１枚ずらす", TARGET
		ENDIF
		IF TEQUIP:振動棒 || TEQUIP:肛用振動棒 || TEQUIP:肛塞
			SIF TEQUIP:振動棒 || TEQUIP:肛用振動棒
				PRINT 振動棒
			SIF TEQUIP:振動棒 && TEQUIP:肛塞
				PRINT 和
			SIF TEQUIP:肛塞
				PRINT 肛塞
			;淫乱
			IF TALENT:淫乱
				PRINT 深深吞入體內的
			ELSE
				PRINT 所貫穿的
			ENDIF
			PRINT 股間
		ELSEIF TALENT:男性
			PRINT 勃起的陰莖
		;胖次を是ていて、ずらしていない
		ELSE
			PRINTFORM 將%TEXTS("阴部：視覚")%
		ENDIF
		PRINT 用灼熱的視線視奸著的同時、
		SETBIT TEQUIP:足交, 1
	ENDIF
	IF PENIS(PLAYER)
		SIF TEQUIP:推倒 == 0
			PRINTFORM %CALLNAME:PLAYER%の
		PRINT 陰莖
	ELSE
		PRINTFORM \@ PREVPLAY("Ｖ挿入", MASTER) ? 被狠狠侵犯過的 # 毫無防備的 \@阴部
	ENDIF
	IF TEQUIP:推倒 == 0 && ABL:施虐属性 >= 3 && RAND:3 == 0
		PRINT 用雙腳
		SETBIT TEQUIP:足交, 2
	;靴を履いている
	ELSEIF EQUIP:靴
		PRINTFORM 用%SHOES(TARGET, "略")%的堅硬底端
	;短襪を是ている
	ELSEIF EQUIP:襪子
		PRINTFORM 用被%SOCKS(TARGET, "略")%包裹著的腳
	;動物足
	ELSEIF TALENT:動物足 && CHOSECOM(SELECTCOM) == 0
		PRINT 用非人的獸腳
	ELSE
		PRINT 用腳
	ENDIF
	IF GETBIT(TEQUIP:足交, 0)
		PRINT 轉動著
	ELSEIF ABL:施虐属性 >= 3 && ABL:技巧 >= 3 && RAND:2 == 0
		PRINT 技巧嫻熟的
		SIF RAND:2
			PRINT 不斷變化蹂躪節奏的
	ELSEIF ABL:施虐属性 >= 3
		PRINT 像是在享受痛苦的反應一樣粗暴地
	ELSEIF ABL:技巧 >= 3 && PENIS(PLAYER)
		PRINT 扑哧扑哧地
	ELSEIF ABL:施虐属性 < 3 || ABL:技巧 < 3
		PRINT 笨拙而小心的
	ENDIF
	IF PREVCOM == SELECTCOM && GETBIT(TEQUIP:足交, 0) && COND("前Ｔ調教者") == PLAYER
		PRINT 持續猛力踐踏著
	ELSEIF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
		IF PENIS(PLAYER)
			PRINT 持續擼動著
		ELSE
			PRINT 持續蹂躪著
		ENDIF
	ELSEIF GETBIT(TEQUIP:足交, 0)
		PRINT 開始猛烈地踐踏起來
	ELSE
		IF PENIS(PLAYER)
			PRINT 擼動起來
		ELSE
			PRINT 蹂躪起來
		ENDIF
	ENDIF
	PRINTW …
ENDIF

@COM_ORDER_COM95, ARG
#DIM MEMO_STAIN
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:施虐属性
IF ABL:施虐属性
	LOCAL += MIN(ABL:施虐属性, 5) * 5
	LOCALS = %LOCALS% + 施虐属性LV{ABL:施虐属性}({MIN(ABL:施虐属性, 5) * 5})
ENDIF
;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF
;ABL:精液中毒
IF ABL:精液中毒
	LOCAL += MIN(ABL:精液中毒, 5) * 3
	LOCALS = %LOCALS% + 精液中毒LV{ABL:精液中毒}({MIN(ABL:精液中毒, 5) * 3})
ENDIF
;強気
IF TALENT:強気
	LOCAL += 10
	LOCALS = %LOCALS% + 強気(10)
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 5
	LOCALS = %LOCALS% - 害羞(5)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF

IF PENIS(PLAYER)
	MEMO_STAIN = COM_ORDER_STAIN("Ｐ", "脚") / 5
ELSE
	MEMO_STAIN = COM_ORDER_STAIN("Ｖ", "脚") / 5
ENDIF
;汚れ有
IF MEMO_STAIN
	LOCAL -= MEMO_STAIN
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({MEMO_STAIN})
ENDIF

;合計を表示(30以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 30

RETURN RESULT


code:
@COM96
;肛門奉仕
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 96

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;-------------------------------------------------
;射精格子
;-------------------------------------------------
IF CHECK_COM("玩弄Ａ手交")
	CALL SAMEN_CHECK, 70
	SELECTCOM = PREVCOM
	CALL SAMEN_CHECK, 50
	SELECTCOM = 96
ELSE
	CALL SAMEN_CHECK
ENDIF

RETURN 1

@FLAG_COM96, ARG
SIF TEQUIP:推倒 == 0
	CALL SETFLAG, "Ａ奉仕FLAG", TARGET
SIF CHECK_COM("玩弄Ａ手交") == 96
	CALL FLAG_COM90, 50
;対象の口⇔調教者のA的污垢が移動
CALL EQ_STAIN, TARGET, "Ｍ", PLAYER, "Ａ"

@SOURCE_COM96, ARG
SIF CHECK_COM("玩弄Ａ手交") == 96
	CALL SOURCE_COM90, 50

CALL VARSET_COMF

CALL SOURCE_SERVICE_PLAY, "肛門奉仕"

CALL SOURCE_DOWNBASE_HP, 96

SOURCE:0:不潔 = COM_ORDER_STAIN("Ａ", "Ｍ")*100 + 100
SOURCE:0:欲情追加 = CALCF("実行値による欲情追加")
SOURCE:0:逸脱 = 3000

IF TEQUIP:推倒
	SOURCE:0:性行動 += 1000
ELSE
	SOURCE:0:屈従 = 5000
ENDIF

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM96
;同時推倒
IF TEQUIP:Ｗ推倒
	PRINTFORM %CALLNAME:TARGET%
	;助手が前回と同じ行動(違う行動の時は、ここでなく、EQUIP_COM32で記述)
	SIF SELECTCOM:1 == PREVCOM:1 && SELECTCOM:1 >= 0
		PRINTFORM %TEXTS("Ｗ推倒中")%、
	IF SELECTCOM == PREVCOM
		PRINTFORMW 將%CALLNAME:MASTER%的肛門持續用舌頭舔舐著…
	ELSE
		IF ABL:奉仕精神 >= 3 && ABL:施虐属性 >= 3 && RAND:2 == 0
			PRINTFORMW 將%CALLNAME:MASTER%的尻穴用舌尖深入的舔舐、摳挖著內側……
		ELSEIF ABL:奉仕精神 >= 3
			PRINTFORMW 將%CALLNAME:MASTER%的肛門用舌頭細緻地舔舐清理著……
			STAIN:MASTER:Ａ = SET_STAIN(0, "Ａ")
		ELSE
			PRINTFORMW 將%CALLNAME:MASTER%的尻穴用舌頭咕嚕咕嚕的玩弄著……
		ENDIF
	ENDIF
ELSEIF CHECK_COM("３Ｐ奉仕")
	CALL TRAIN_MESSAGE_3P
ELSE
	PRINTFORM %CALLNAME:TARGET%
	;前回逆肛交
	IF PREVPLAY("Ａ挿入", PLAYER)
		PRINTFORM 將失去填充物而不斷張合著的%CALLNAME:PLAYER%的肛門
		IF ABL:奉仕精神 >= 3 && RAND:2 == 0
			PRINTW 用舌頭撫慰著……
		ELSE
			PRINTW 用舌頭舔舐著……
		ENDIF
	ELSE
		IF CHECK_COM("玩弄Ａ手交")
			IF PREVCOM == 97
				LOCALS = %CLO("内裤冲")%
				LOCALS:1 = %LOCALS%
			ELSE
				LOCALS = 手
				LOCALS:1 = 掌
			ENDIF
			IF TALENT:恋慕 && RAND:4 == 0
				PRINTFORM 將最愛的人的陰莖用%LOCALS%溫柔細緻地侍奉著的同時、
			ELSEIF ABL:技巧 >= 3 && ABL:奉仕精神 >= 3 && RAND:3 == 0
				PRINTFORM 將敏感的龜頭用%LOCALS:1%來回地揉搓旋轉著的同時、
			ELSEIF ABL:技巧 >= 3 && RAND:2 == 0
				PRINTFORM 將陰莖用絕妙的力道刺激著的同時、
			ELSEIF RAND:2 == 0
				PRINTFORM 將陰莖用%LOCALS%慢慢地擼動著的同時、
			ELSE
				PRINTFORM 將陰莖用%LOCALS%擼動著的同時、
			ENDIF
		ENDIF
		IF ABL:奉仕精神 >= 3 && ABL:施虐属性 >= 3 && RAND:2 == 0
			PRINTFORMW 將%CALLNAME:MASTER%的尻穴用舌尖深入的舔舐、摳挖著內側……
		ELSEIF ABL:奉仕精神 >= 3
			PRINTFORMW 細緻地用舌頭舔舐着肛門清理著……
			STAIN:PLAYER:Ａ = SET_STAIN(0, "Ａ")
		ELSEIF ABL:施虐属性 >= 3 || TEQUIP:推倒
			PRINTFORMW 將%CALLNAME:MASTER%的尻穴用舌頭咕嚕咕嚕的玩弄著……
		ELSE
			PRINTFORMW 將%CALLNAME:MASTER%的肛門戰戰兢兢地舔舐著……
		ENDIF
	ENDIF
ENDIF

@COM_ORDER_COM96, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF

;害羞
IF TALENT:害羞
	LOCAL -= 5
	LOCALS = %LOCALS% - 害羞(5)
ENDIF
;献身的
IF TALENT:献身
	LOCAL += 6
	LOCALS = %LOCALS% + 献身(6)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF

;汚れ有
IF COM_ORDER_STAIN("Ａ", "Ｍ")
	LOCAL -= COM_ORDER_STAIN("Ａ", "Ｍ")
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({COM_ORDER_STAIN("Ａ", "Ｍ")})
ENDIF

;合計を表示(40以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 50

RETURN RESULT


code:
@COM97
;内褲冲
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 97

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;-------------------------------------------------
;射精格子
;-------------------------------------------------
IF CHECK_COM("玩弄Ａ手交")
	CALL SAMEN_CHECK, 70
	SELECTCOM = 96
	CALL SAMEN_CHECK, 50
	SELECTCOM = 97
ELSEIF CHECK_COM("洗面奶手交")
	CALL SAMEN_CHECK, 70
	SELECTCOM = 104
	CALL SAMEN_CHECK, 50
	SELECTCOM = 97
ELSE
	CALL SAMEN_CHECK
ENDIF

RETURN 1

@FLAG_COM97, ARG
IF CHECK_COM("玩弄Ａ手交") == 97
	CALL FLAG_COM96, 50
ELSEIF CHECK_COM("洗面奶手交") == 97
	CALL FLAG_COM104, 50
ENDIF
;対象の胖次⇔調教者のＰ的污垢が移動
CALL EQ_STAIN, TARGET, "胖次", PLAYER, "Ｐ"

;汚れは胖次でふき取る。
CALL RESET_STAIN_EX, "Ｐ", PLAYER
;布鲁马と胖次はズレなどを戻す
FOR LOCAL, 60, 64
	CLEARBIT EQUIP:胖次, LOCAL
	CLEARBIT EQUIP:外套, LOCAL
NEXT

@SOURCE_COM97, ARG
IF CHECK_COM("玩弄Ａ手交") == 97
	CALL SOURCE_COM96, 50
ELSEIF CHECK_COM("洗面奶手交") == 97
	CALL SOURCE_COM104, 50
ENDIF

CALL VARSET_COMF

CALL SOURCE_SERVICE_PLAY, "内褲冲"

CALL SOURCE_DOWNBASE_HP, 97

SOURCE:0:不潔 = COM_ORDER_STAIN("Ｐ", "手")*10
SOURCE:0:欲情追加 = CALCF("実行値による欲情追加")
SOURCE:0:屈従 = 500
SOURCE:0:逸脱 = 100

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM97
IF CHECK_COM("Ｗ奉仕")
	;二人とも前回と同じ行動
	IF SELECTCOM:1 == PREVCOM:1 && SELECTCOM == PREVCOM
		PRINTFORM 兩人
		IF TALENT:(TARGET:1):淫乱 + TALENT:(TARGET:1):恋慕 && TALENT:淫乱 + TALENT:恋慕 && RAND:3 == 0
			PRINT 神情恍惚、眼神濕潤的對視著的同時
		ELSEIF RAND:2 == 0
			PRINT 如同完全沉浸於快感一樣無意識的漏出涎水的同時
		ENDIF
		PRINTFORMW 將%CALLNAME:MASTER%的陰莖用帶著溫熱的胖次包裹著持續擼動著……
	ELSE
		PRINTFORM %CALLNAME:TARGET%和%CALLNAME:(TARGET:1)%
		IF SELECTCOM:1 != PREVCOM:1 && SELECTCOM != PREVCOM && RAND:2 == 0
			PRINTFORMW 像是特意向%CALLNAME:MASTER%展示一樣慢慢的脫下了自己的胖次、按照解下的順序用胖次包裹住陰莖開始擼動起來……
		ELSE
			PRINTFORMW 用兩人的胖次將陰莖包裹住、開始擼動起來……
		ENDIF
	ENDIF
;==90で两根手交、== 91なら口交＆手交、それ以外は犯されながら手交
ELSEIF CHECK_COM("３Ｐ奉仕")
	CALL TRAIN_MESSAGE_3P
;同時推倒
ELSEIF TEQUIP:Ｗ推倒
	PRINTFORM %CALLNAME:TARGET%
	;助手が前回と同じ行動(違う行動の時は、ここでなく、EQUIP_COM32で記述)
	SIF SELECTCOM:1 == PREVCOM:1 && SELECTCOM:1 >= 0
		PRINTFORM 、%TEXTS("Ｗ推倒中")%、

	PRINTFORM 將%CALLNAME:MASTER%的陰莖、用
	IF CHECK_CLO("湿的") && PREVCOM != SELECTCOM
		PRINT 因愛液而浸染出羞恥色斑的
	ELSEIF PREVCOM != SELECTCOM
		PRINT 剛剛脫下、還殘留著熱意的
		IF CHECK_CLO("湿的") && RAND:2 == 0
			PRINT 濡濕的
		ELSEIF CHECK_CLO("闷热的")
			PRINT 蒸騰著熱汽的
		ENDIF
	ENDIF
	PRINTFORM 將%CLO("内裤冲")%
	IF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
		PRINTW でゴシゴシと扱いている…
	ELSE
		PRINTW で包んで扱いてきた…
	ENDIF
ELSE
	PRINTFORM %CALLNAME:TARGET%
	;被逆推ているとき
	IF TEQUIP:面部騎乗 && TEQUIP:推倒
		IF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
			PRINTFORM %CALLNAME:PLAYER%
			IF RAND:3 == 0 && ABL:欲望 + ABL:施虐属性 >= 5
				PRINTFORM 的臉頰用%TEXTS("阴部：感触")%緊緊壓住不肯放開的同時、
			ELSEIF RAND:2 && ABL:欲望 + ABL:施虐属性 >= 3
				PRINT 的臉頰用股間緊密的摩擦著的同時、
			ELSE
				PRINT 的臉頰跨坐著、保持著這樣的姿勢、
			ENDIF
			PRINTFORMW 用%CLO("内裤冲")%將陰莖咕啾咕啾的擼動著……
		ELSE
			IF EQUIP:外套 == 0 && CHECK_PANTIES("繩系胖次") && TALENT:害羞 == 0 && RAND:2 == 0
				PRINTFORM 用豐腴的臀部壓迫著%CALLNAME:PLAYER%的臉頰的同時、
			ELSEIF EQUIP:外套 == 0 && CHECK_PANTIES("繩系胖次")
				PRINTFORM 暫時將股間從%CALLNAME:PLAYER%的臉頰上移開、
			ELSE
				PRINTFORM 保持著跨坐在%CALLNAME:PLAYER%臉頰上的姿勢、
			ENDIF
			PRINTFORM 將%CLO("内裤冲")%
			IF EQUIP:外套 == 0 && CHECK_PANTIES("繩系胖次")
				PRINT 的繫帶解開、身體前傾、用
			ELSE
				PRINTFORM 緩緩脫下、重新將腰肢下沉到臉頰上後、用
			ENDIF
			IF CHECK_CLO("湿的") && RAND:2 == 0
				PRINT 濡濕的
			ELSEIF CHECK_CLO("闷热的")
				PRINT 蒸騰著熱汽的
			ENDIF
			PRINTW 胖次擼動著陰莖…
		ENDIF
	ELSE
		;六九式
		SIF TEQUIP:六九式
			PRINTFORM %TEXTS("６９")%
		IF TEQUIP:接吻
			PRINTFORM %CALLNAME:PLAYER%%TEXTS("接吻継続助詞付き")%、
			CALL SETFLAG, "追加接吻フラグ", TARGET
		ELSEIF CHECK_COM("玩弄Ａ手交")
			PRINTFORM 將%CALLNAME:PLAYER%的尻穴用\@ RAND:2 ? 盡力延長的 # \@舌尖
			IF RAND:3 == 0
				PRINT 蹂躪著的同時、用
			ELSEIF RAND:2
				SIF (ABL:奉仕精神 >= 3 || ABL:施虐属性 >= 3) && RAND:2
					PRINT 連同最深處的肉襞也一併
				PRINTFORM 舔舐著的同時、用
			ELSE
				PRINT 玩弄著的同時、用
			ENDIF
		ELSEIF CHECK_COM("洗面奶手交")
			PRINTFORM 將%CALLNAME:PLAYER%的臉頰用
			IF RAND:3 == 0
				PRINT 豐腴的胸部
			ELSEIF RAND:2 == 0
				PRINT 成熟飽滿的果實
			ELSE
				PRINT 柔軟的乳房
			ENDIF
			IF RAND:2 && TEQUIP:推倒
				PRINTFORM 不斷壓迫著的同時、用
			ELSEIF RAND:2
				PRINTFORM 緊緊包裹住的同時、用
			ELSE
				PRINTFORM 的谷間深深埋住的同時、用
			ENDIF
		ELSEIF INSERT("合体中")
			PRINTFORM %TEXTS("合体中", TARGET)%
		ELSEIF CHECK_CLO("湿的") && PREVCOM != SELECTCOM
			PRINT 被愛液浸濕的
		ELSEIF PREVCOM != SELECTCOM
			PRINT 剛剛脫下、還殘留著熱意的
			IF CHECK_CLO("湿的") && RAND:2 == 0
				PRINT 濡濕的
			ELSEIF CHECK_CLO("闷热的")
				PRINT 蒸騰著熱汽的
			ENDIF
		ENDIF
		PRINTFORM 將%CLO("内裤冲")%
		IF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
			PRINTW 將陰莖噗呲噗呲的擼動起來……
		ELSEIF RAND:3 == 0 && (TALENT:害羞 == 0 || TALENT:淫乱 || TALENT:恋慕)
			IF RAND:2 || CHECK_PANTIES("兜襠布")
				PRINTW 與陰部接觸的部分將龜頭摩擦著……
			ELSE
				PRINTW 的濡濕部分包裹住龜頭、在將透過布料溢出的汁液舔入口中的同時、也不斷地擼動著莖身……
			ENDIF
		ELSEIF RAND:2 == 0
			PRINTW 將陰莖整個包裹住、擼動起來……
		ELSE
			PRINTW 將陰莖的上端包裹住、噗嗤噗嗤的擼動起來……
		ENDIF
	ENDIF
ENDIF
;母乳体質とかだと母乳を飲めるかも
CALL TEXT, "喝母乳"

@COM_ORDER_COM97, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF
;ABL:精液中毒
IF ABL:精液中毒
	LOCAL += MIN(ABL:精液中毒, 5) * 2
	LOCALS = %LOCALS% + 精液中毒LV{ABL:精液中毒}({MIN(ABL:精液中毒, 5) * 2})
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 3
	LOCALS = %LOCALS% - 害羞(3)
ENDIF
;献身的
IF TALENT:献身
	LOCAL += 6
	LOCALS = %LOCALS% + 献身(6)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF

;汚れ有
IF COM_ORDER_STAIN("Ｐ", "手")/3
	LOCAL -= COM_ORDER_STAIN("Ｐ", "手")/3
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({COM_ORDER_STAIN("Ｐ", "手")/3})
ENDIF

;合計を表示(20以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 20

RETURN RESULT


code:
@COM98
;奉仕舐陰
CALL PRINT_TRAIN_NAME(SELECTCOM)
;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 98

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL SETFLAG, "下半身を脱がす", PLAYER

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;-------------------------------------------------
;射精格子
;-------------------------------------------------
CALL SAMEN_CHECK

RETURN 1

@FLAG_COM98, ARG
;対象の口⇔調教者のＶ的污垢が移動
CALL EQ_STAIN, TARGET, "Ｍ", PLAYER, "Ｖ"
SIF CHECK_COM("清洁舔阴", 98)
	CALL SETFLAG, "清洁舔阴", TARGET

@SOURCE_COM98, ARG
CALL VARSET_COMF

CALL SOURCE_SERVICE_PLAY, "奉仕舐陰"

CALL SOURCE_DOWNBASE_HP, 98

SOURCE:0:不潔 = COM_ORDER_STAIN("Ｖ", "Ｍ")*100 + 100
SOURCE:0:欲情追加 = CALCF("実行値による欲情追加")
SOURCE:0:屈従 = 1000
SOURCE:0:逸脱 = 500

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM98
;同時推倒
IF TEQUIP:Ｗ推倒
	PRINTFORM %CALLNAME:TARGET%
	;助手が前回と同じ行動(違う行動の時は、ここでなく、EQUIP_COM32で記述)
	SIF SELECTCOM:1 == PREVCOM:1 && SELECTCOM:1 >= 0
		PRINTFORM 、%TEXTS("Ｗ推倒中")%、

	IF PREVCOM == SELECTCOM
		PRINTFORMW 將%CALLNAME:PLAYER%的陰部用指尖慢慢展開、仔細地持續舔舐著……
	ELSEIF TEQUIP:推倒
		;前回逆強姦
		IF PREVPLAY("Ｖ挿入", PLAYER)
			IF PREVEX("射精") && RAND:3
				PRINTFORM 將正在噗嚕噗嚕的溢出灼熱精液的
			ELSEIF ABL:施虐属性 >= 3 && RAND:2 == 0
				PRINTFORM 將被盡情的蹂躪過的
			ELSE
				PRINTFORM 將因為之前的激烈侵犯而微微顫抖的
			ENDIF
		ELSEIF RAND:3 == 0
			PRINTFORM 身體被牢牢禁錮住、動彈不得的
		ELSE
			PRINTFORM 將被牢牢壓住的
		ENDIF
		PRINTFORMW %CALLNAME:PLAYER%的陰唇用舌頭仔細地舔舐進攻著……
	ENDIF
ELSEIF CHECK_COM("３Ｐ奉仕")
	CALL TRAIN_MESSAGE_3P
ELSEIF CHECK_COM("恥垢清洁舔阴")
	PRINTFORM %CALLNAME:TARGET%將%CALLNAME:PLAYER%壓住、將試圖掩蓋的雙手撥開、將散發著微妙氣味的陰部展開、
	PRINTFORM 將%TEXTR("陰唇內側殘留著的恥垢/膣口沒能清潔乾淨的頑固污垢/濁白色的骯髒恥垢/黏稠、猶如奶酪一樣的污垢/粘膜上沾染的羞恥污垢")%
	IF RAND:2 && PLAYER == MASTER && TALENT:气味弱点 && TALENT:恋慕
		PRINTFORMW 的%TEXTS("气味")%嗅聞著的同時、用靈活的唇舌將所有恥垢都徹底的舔舐乾淨了……
	ELSEIF RAND:2
		PRINTFORMW 用靈巧的舌頭將每一寸角落都細緻的舔舐刮取了……
	ELSE
		PRINTFORMW 用伸長的舌尖將整片陰部都漂亮的掃除乾淨了……
	ENDIF
ELSEIF PREVCOM == SELECTCOM
	PRINTFORMW %CALLNAME:TARGET%將%CALLNAME:PLAYER%的陰部用指尖慢慢展開、仔細地持續舔舐著……
ELSEIF TEQUIP:推倒
	PRINTFORM %CALLNAME:TARGET%
	;前回逆強姦
	IF PREVPLAY("Ｖ挿入", PLAYER)
		IF PREVEX("射精") && CHECK_STAIN(STAIN:PLAYER:Ｖ, "精液") && RAND:3
			PRINTFORM 將正在噗嚕噗嚕的溢出灼熱精液的
		ELSEIF ABL:施虐属性 >= 3 && RAND:2 == 0
			PRINTFORM 將被盡情的蹂躪過的
		ELSE
			PRINTFORM 將因為之前的激烈侵犯而微微顫抖的
		ENDIF
	ELSEIF RAND:3 == 0
		PRINTFORM 身體被牢牢禁錮住、動彈不得的
	ELSE
		PRINTFORM 將被牢牢壓住的
	ENDIF
	IF CHECK_COM("清洁舔阴")
		IF RAND:2
			PRINTFORMW %CALLNAME:PLAYER%的陰部被展開、殘留的濃臭污垢被靈巧的舌頭漂亮的掃除乾淨了……
		ELSE
			PRINTFORMW %CALLNAME:PLAYER%的膣口被伸長的舌尖插入、借機吮吸起內部溢出的、甜膩愛液與被化開稀釋的污垢所混合成的汁液……
		ENDIF
	ELSE
		PRINTFORMW %CALLNAME:PLAYER%的陰唇用舌頭仔細地舔舐進攻著……
	ENDIF
ELSE
	PRINTFORM %CALLNAME:TARGET%
	;尿のお掃除
	IF CHECK_COM("おしっこ清洁舔阴")
		IF PREVPLAY("膣内浴尿", PLAYER)
			PRINTFORM 漸漸逆流出之前注入的尿液和體內愛液的混合粘液的
		ELSE
			PRINTFORM 排尿之後散發著%TEXTR("嗆鼻的/強烈的/氨水的")%劇烈臭味的
		ENDIF
	ELSEIF PREVPLAY("Ｖ挿入", PLAYER) || CHECK_COM("清洁舔阴")
		IF PREVEX("Ｖ中出し", PLAYER) && CHECK_STAIN(STAIN:PLAYER:Ｖ, "精液") && RAND:2
			PRINTFORM 漸漸溢出之前灌入膣道的精液的
		ELSEIF RAND:2
			PRINTFORM 因為之前的激烈侵犯而微微顫抖著的
		ELSE
			PRINTFORM 不斷垂流下因激烈攪拌而更加黏稠起泡的淫液的
		ENDIF
	ELSEIF TEQUIP:六九式
		PRINTFORM %TEXTS("６９")%
	ELSEIF EXP_GAIN("絶頂経験", PLAYER)
		PRINTFORM 因為之前激烈絕頂而恍惚著的
	ELSE
		PRINTFORM %TEXTR("恳求口舌侍奉/请求侍奉/在眼前緊張地張開雙腿的")%
	ENDIF

 	IF CHOSECOM(SELECTCOM)
		PRINTFORM %CALLNAME:PLAYER%的%TEXTS("割れ目：表現短め", PLAYER)%
	ELSE
		PRINTFORM %CALLNAME:PLAYER%の%TEXTS("割れ目", PLAYER)%
	ENDIF
	IF COND("指の使用") && TCVAR:干劲 > 20
		PRINTFORM 用指尖慢慢展開、
	ELSE
		PRINTFORM 用臉頰慢慢接近、
	ENDIF
	IF CHECK_COM("清洁舔阴")
		PRINTFORMW 開始熱情的掃除起來……
	ELSEIF TCVAR:干劲 > 20
		IF RAND:4 == 0
			PRINTFORMW 將陰蒂剝開後用力吮吸著……
		ELSEIF RAND:3 == 0
			PRINTFORMW 與膣口相吻、執拗地刺激著……
		ELSEIF RAND:2
			PRINTFORMW 將蠕動肉襞的每一處褶皺用舌尖舒展開來……
		ELSE
			PRINTFORMW 沉浸於淫靡的氛圍中、啾、啾的熱情啄吻著……
		ENDIF
	ELSEIF RAND:2
		PRINTFORMW 用舌頭侍奉起來……
	ELSE
		PRINTFORMW \@ (ABL:信頼 >= 2 && RAND:2) ? 順從地 # 呸嘍呸嘍地 \@舔舐起來……
	ENDIF

	SIF CHOSECOM(SELECTCOM) == 0 && TALENT:男性 == 0 && COND("性別嫌悪") && CHECK_COM("清洁舔阴") == 0
		PRINTFORMW 像是對將同性的性器用嘴侍奉有著抵觸心一樣、露出了難以形容的複雜神色……
ENDIF

@COM_ORDER_COM98, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF

;害羞
IF TALENT:害羞
	LOCAL -= 3
	LOCALS = %LOCALS% - 害羞(3)
ENDIF
;献身的
IF TALENT:献身
	LOCAL += 6
	LOCALS = %LOCALS% + 献身(6)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF

;汚れ有
IF COM_ORDER_STAIN("Ｖ", "Ｍ")
	LOCAL -= COM_ORDER_STAIN("Ｖ", "Ｍ")
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({COM_ORDER_STAIN("Ｖ", "Ｍ")})
ENDIF
;男性
IF TALENT:男性
	LOCAL += 10
	LOCALS = %LOCALS% + 男性(10)
ENDIF

;合計を表示(30以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 30

RETURN RESULT


code:
@COM999
;口上側からの調教強制終了用
BEGIN AFTERTRAIN


code:
@COM99
;六九式
CALL PRINT_TRAIN_NAME(SELECTCOM)

IF TEQUIP:六九式
	PRINTL
	PRINTFORMW \@ TEQUIP:推倒 ? %CALLNAME:TARGET% # %CALLNAME:PLAYER% \@は六九式の体勢をやめた…
	STR:前回指令 = %STR:今回指令%
	TEQUIP:六九式 = 0
	TEQUIP:PLAYER:六九式 = 0
	PREVCOM = 99
	RETURN 0
ENDIF
;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 99

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
;相互お掃除PLAY
IF CHECK_COM("清洁口交", TNUM("口交")) + CHECK_COM("清洁舔阴", TNUM("奉仕舐陰")) && CHECK_COM("清洁口交", TNUM("舐陰莖")) + CHECK_COM("清洁舔阴", TNUM("舐陰"))
	IF CHECK_COM("清洁口交", TNUM("口交"))
		TEQUIP:六九式 = TNUM("口交")
	ELSE
		TEQUIP:六九式 = TNUM("奉仕舐陰")
	ENDIF
	IF CHECK_COM("清洁口交", TNUM("舐陰莖"))
		TEQUIP:PLAYER:六九式 = TNUM("舐陰莖")
	ELSE
		TEQUIP:PLAYER:六九式 = TNUM("舐陰")
	ENDIF
ELSE
	;阴茎あるならそっち優先
	IF PENIS(PLAYER)
		TEQUIP:六九式 = TNUM("口交")
	ELSE
		TEQUIP:六九式 = TNUM("奉仕舐陰")
	ENDIF
	IF PENIS(TARGET)
		TEQUIP:PLAYER:六九式 = TNUM("舐陰莖")
	ELSE
		TEQUIP:PLAYER:六九式 = TNUM("舐陰")
	ENDIF
ENDIF

CALL TRAIN_MESSAGE_B

IF REFUSE_CHECK()
	TEQUIP:六九式 = 0
	TEQUIP:PLAYER:六九式 = 0
	RETURN 0
ENDIF

STR:前回指令 = %STR:今回指令%

;面部騎乗ＯＮ
SIF TEQUIP:面部騎乗 + TEQUIP:推倒 && TEQUIP:PLAYER:六九式 == TNUM("舐陰")
	TEQUIP:面部騎乗 = 1
;-------------------------------------------------
;射精格子
;-------------------------------------------------
CALL SAMEN_CHECK

RETURN 1

@FLAG_COM99, ARG
TRYCALLFORM FLAG_COM{TEQUIP:PLAYER:六九式}, 70
TRYCALLFORM FLAG_COM{TEQUIP:六九式}, 70

@SOURCE_COM99, ARG
TRYCALLFORM SOURCE_COM{TEQUIP:PLAYER:六九式}, 70
TRYCALLFORM SOURCE_COM{TEQUIP:六九式}, 70

CALL VARSET_COMF

CALL SOURCE_SERVICE_PLAY, "六九式"

SOURCE:0:欲情追加 = CALCF("実行値による欲情追加") + 500
SOURCE:0:露出 = 1400
SOURCE:0:屈従 = 1300
SOURCE:0:逸脱 = 800

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM99
;相互お掃除PLAY
IF CHECK_COM("清洁口交", TNUM("口交")) + CHECK_COM("清洁舔阴", TNUM("奉仕舐陰")) && CHECK_COM("清洁口交", TNUM("舐陰莖")) + CHECK_COM("清洁舔阴", TNUM("舐陰"))
	PRINTFORMW %CALLNAME:PLAYER%與%CALLNAME:TARGET%將彼此性器上的淫靡污垢用唇舌打掃乾淨了……
;推倒
ELSEIF TEQUIP:推倒
	IF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
		PRINTFORMW %CALLNAME:TARGET%將由自己主導的六九式口交盡情的享受著……
	ELSEIF PREVCOM == TNUM("口交") && COND("前Ｔ調教者") == PLAYER
		PRINTFORM 在%CALLNAME:TARGET%將挺立的陰莖含住吮吸著的同時、跨坐在%CALLNAME:PLAYER%的臉頰上、
		IF TEQUIP:PLAYER:六九式 == TNUM("舐陰莖")
			PRINTFORMW 焦急地催促對方盡心舔舐眼前的蜜泉……
		ELSE
			PRINTFORMW 迫不及待地將陰部直直地印在臉頰上……
		ENDIF
	ELSE
		IF TEQUIP:面部騎乗 && COND("前Ｔ調教者") == PLAYER
			PRINTFORM %CALLNAME:TARGET%將%CALLNAME:PLAYER%的臉頰用股間壓迫、騎乘著的同時、前傾身體
		ELSE
			PRINTFORM %CALLNAME:TARGET%將%CALLNAME:PLAYER%\@ COND("脚の使用") ? 越過、將 # 的 \@
			IF TEQUIP:PLAYER:六九式 == TNUM("舐陰莖")
				PRINTFORM 口腔用下體不斷侵犯著的同時
				SIF TEQUIP:六九式 == TNUM("口交")
					PRINTFORM \@ COND("脚の使用") ?  # 、 \@自己也
			ELSE
				PRINTFORM 臉頰用股間激烈摩擦著的同時
				SIF TEQUIP:六九式 == TNUM("奉仕舐陰")
					PRINTFORM \@ COND("脚の使用") ?  # 、 \@自己也
			ENDIF
		ENDIF
		IF TEQUIP:六九式 == TNUM("口交")
			PRINTFORMW 陰莖投入的吮吸著……
		ELSE
			PRINTFORMW 將陰部用舌頭反復舔舐著……
		ENDIF
	ENDIF
ELSE
	IF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
		PRINTFORMW %CALLNAME:TARGET%與%CALLNAME:PLAYER%將彼此的%TEXTR("性器仔細地舔舐著/股間用臉頰填滿、將內部的秘處盡情地品味著")%……
	ELSE
		IF COND("脚の使用") == 0
			PRINTFORM %CALLNAME:PLAYER%與%CALLNAME:TARGET%轉動身體、形成六九式的姿勢、
		ELSE
			PRINTFORM %CALLNAME:PLAYER%任由%CALLNAME:TARGET%跨坐在臉頰上、
		ENDIF
		IF PENIS(PLAYER) == PENIS(TARGET)
			PRINTFORMW 將彼此的\@ PENIS(PLAYER) ? 陰莖吮吸、侍奉著 # 陰部仔細地舔舐著 \@…
		ELSE
			IF TEQUIP:六九式 == TNUM("口交")
				PRINTFORM 在勃起的陰莖享受著溫熱而濕潤的口腔的殷切侍奉的同時、
			ELSE
				PRINTFORM 在微顫的陰部與不斷張合的雙唇接觸的邊角流出涎水與蜜液的混合液體的同時、
			ENDIF
			IF TEQUIP:PLAYER:六九式 == TNUM("舐陰莖")
				PRINTFORMW 也將眼前蒸騰著醉人熱汽的肉棒仔細品味著……
			ELSE
				PRINTFORMW 也將不斷散發出淫靡%TEXTS("气味")%的陰部用舌頭細心舔舐著……
			ENDIF
		ENDIF
	ENDIF
ENDIF

@SET_PLAY69_TURNEND
;道具指令等は、連続するのが自然な他の指令にする
SELECTCASE TEQUIP:PLAYER:六九式
CASE TNUM("振動棒")
	TEQUIP:PLAYER:六九式 = 8
CASE TNUM("肛用振動棒"), TNUM("肛塞")
	TEQUIP:PLAYER:六九式 = 2
CASE 141
	TEQUIP:PLAYER:六九式 = 8
CASE 142
	TEQUIP:PLAYER:六九式 = 2
CASE 143
	IF RAND:2
		TEQUIP:PLAYER:六九式 = 2
	ELSE
		TEQUIP:PLAYER:六九式 = 8
	ENDIF
ENDSELECT

@SET_PLAY69
;指令連続選択による派生
CALL COMABLE_69
;前回と今回の調教者が違うorＷ推倒中or六九式不可
IF RESULT == 0 || COND("前Ｔ調教者") != PLAYER || TEQUIP:Ｗ推倒
	TEQUIP:六九式 = 0
	TEQUIP:PLAYER:六九式 = 0
	RETURN 0
ELSE
	;褒めた時など、体勢が変わらない指令
	SELECTCASE SELECTCOM
	;推倒状態の媚薬は騎乗位に移行するので解除
	CASE 261
		IF TEQUIP:推倒
			TEQUIP:六九式 = 0
			TEQUIP:PLAYER:六九式 = 0
		ENDIF
	ENDSELECT
ENDIF

SELECTCASE CHECK_EQUIP_69(SELECTCOM)
CASE 1
	;前回の調教が口交、舐陰強制か六九式だと六九式に
	IF PREVCOM == TNUM("口交") || PREVCOM == TNUM("奉仕舐陰") || PREVCOM == TNUM("深喉") || TEQUIP:六九式
		SIF TEQUIP:六九式 == 0
			TEQUIP:六九式 = PREVCOM
		TEQUIP:PLAYER:六九式 = SELECTCOM
	ENDIF
CASE 2
	;前回の調教が舐陰、舐陰莖か六九式だと六九式に
	IF PREVCOM == TNUM("舐陰") || PREVCOM == TNUM("舐陰莖") || TEQUIP:面部騎乗 || TEQUIP:六九式
		;69に移行した時（TEQUIP:PLAYER:六九式は0も意味を持つので、TEQUIP:六九式で判定。）
		IF TEQUIP:六九式 == 0
			IF TEQUIP:面部騎乗
				TEQUIP:PLAYER:六九式 = TEQUIP:面部騎乗
			ELSE
				TEQUIP:PLAYER:六九式 = PREVCOM
			ENDIF
		ENDIF
		TEQUIP:六九式 = SELECTCOM
	ENDIF
CASE 3
	;六九式中だと継続
	IF TEQUIP:六九式
		TEQUIP:PLAYER:六九式 = SELECTCOM
		;partnerの行動に変化をつける
		IF RAND:3 == 0 && PENIS(PLAYER)
			IF CANTRAIN("乳交") && RAND:3 == 0
				TEQUIP:六九式 = TNUM("乳交")
			ELSEIF CANTRAIN("口交") && RAND:2
				TEQUIP:六九式 = TNUM("口交")
			ELSEIF CANTRAIN("手淫")
				TEQUIP:六九式 = TNUM("手淫")
			ENDIF
		ENDIF
	ENDIF
CASE 4
	;六九式中だと継続
	SIF TEQUIP:六九式
		TEQUIP:六九式 = SELECTCOM
;六九式も可能な指令(TEQUIP:六九式は変更させない)
CASE 6
	RETURN 0
CASEELSE
	IF SELECTCOM != 99
		TEQUIP:六九式 = 0
		TEQUIP:PLAYER:六九式 = 0
	ENDIF
ENDSELECT

@CHECK_EQUIP_69(ARG)
#FUNCTION

;時間を経過させない指令なら変化无
SIF NOTTIME_PLAY(ARG)
	RETURNF 6

SELECTCASE ARG
;PLAYERの行動　六九式に移行する股間弄り指令
CASE TNUM("舐陰"), TNUM("舐陰莖")
	RETURNF 1
;TARGETの行動　六九式に移行する奉仕指令
CASE TNUM("口交"), TNUM("奉仕舐陰"), TNUM("深喉")
	RETURNF 2
;PLAYERの行動　六九式体勢で可能な愛撫系の股間弄り指令
CASE 0 TO 2, 7 TO 10, 15
	RETURNF 3
;PLAYERの行動　道具指令や打屁股、拳交
CASE 30 TO 39, 130, 141 TO 143
	RETURNF 3
;TARGETの行動　六九式体勢で可能な奉仕指令
CASE TNUM("手淫") TO TNUM("乳交"), TNUM("内褲冲"), TNUM("奉仕舐陰"), TNUM("飞机杯冲")
	RETURNF 4
;六九式も可能な指令(TEQUIP:六九式は変更させない)
CASE 3, 13, 14, 24, 146, 176, 230 TO 300
	RETURNF 6
CASE 99
	RETURNF 7
ENDSELECT
RETURNF 0

@COMABLE_69
;六九式実行判定
SIF TEQUIP:六九式
	RETURN 1
;拘束（足or体勢変更不可）だとダメ
SIF GETBITS(TEQUIP:拘束, 21, 22)
	RETURN 0

;実行条件判定
CALL COM_ORDER_COM99, 1
RETURN RESULT

@EQUIP_COM69
SIF TEQUIP:六九式 == 0
	RETURN 0
;アクティブでない方の動きによるSOURCE等の変化を記述
SIF SELECTCOM == 99
	RETURN 0

SELECTCASE SELECTCOM
;６９自慰中
CASE 3
	CALLFORM SOURCE_COM{TEQUIP:六九式}, 50
	CALLFORM FLAG_COM{TEQUIP:六九式}, 50
;PLAYERの行動　舐陰or舐陰莖
CASE 0 TO 2, TNUM("舐陰莖"), 7 TO 10, 15, 30 TO 39, 130, 141 TO 143, TNUM("深喉")
	CALLFORM SOURCE_COM{TEQUIP:六九式}, 50
	CALLFORM FLAG_COM{TEQUIP:六九式}, 50
;TARGETの行動　口交or乳交or奉仕舐陰
CASE TNUM("手淫") TO TNUM("乳交"), TNUM("内褲冲"), TNUM("奉仕舐陰"), TNUM("飞机杯冲"), TNUM("深喉")
	CALLFORM SOURCE_COM{TEQUIP:PLAYER:六九式}, 50
	CALLFORM FLAG_COM{TEQUIP:PLAYER:六九式}, 50
ENDSELECT

@EQUIP_COM69_2
#DIM MEMO_SELECTCOM
SIF TEQUIP:六九式 == 0
	RETURN 0
;アクティブでない方の動きを描写する
SIF SELECTCOM == 99
	RETURN 0
PRINTL ＜六九式中＞

MEMO_SELECTCOM = SELECTCOM

SELECTCASE SELECTCOM
;６９自慰中
CASE 3
	SELECTCOM = TEQUIP:六九式
	CALL GET_EXP_COM

	;射精ゲージの増加
	CALL SAMEN_CHECK, LOCAL
;PLAYERの行動　舐陰or舐陰莖
CASE 0 TO 2, TNUM("舐陰莖"), 7 TO 10, 15, 30 TO 39, 130, 141 TO 143, TNUM("深喉")
	PRINTFORM %CALLNAME:TARGET%

	SELECTCASE SELECTCOM
	CASE 0
		LOCALS = 秘處
	CASE 2, 9, TNUM("肛用振動棒")
		LOCALS = 肛門
	CASE TNUM("舐陰莖"), 10
		LOCALS = 陰莖
	CASE 8
		LOCALS = 敏感地帶
	CASE 15
		LOCALS = 尿尿的穴
	CASE 30
		LOCALS = 陰蒂
	CASE TNUM("振動棒")
		LOCALS = 膣
	CASEELSE
		LOCALS = 陰部
	ENDSELECT

	LOCAL = 50
	IF TEQUIP:推倒
		IF SELECTCOM == TNUM("舐陰")
			IF CHECK_CLO("Ｖ露出")
				PRINT 將臉頰用陰部反復摩擦的同時、
			ELSE
				PRINT 將臉頰用重要的部分緊緊壓住的同時、
			ENDIF
		ELSE
			PRINT 將下體傳來的快樂愈加飢渴的索取著的同時、
		ENDIF
	ELSE
		IF SELECTCOM == 130
			PRINTFORM 將反復拍打在尻部的巴掌
		ELSEIF SELECTCOM >= 141 && SELECTCOM <= 143
			PRINTFORM 將突然擊打在腹部的拳頭
		ELSEIF SELECTCOM == TNUM("振動棒") || SELECTCOM == TNUM("肛用振動棒")
			PRINTFORM 將在%LOCALS%內部反復翻攪的振動棒所帶來的甘美感
		ELSE
			PRINTFORM 將從%LOCALS%襲來的快感
		ENDIF
		;快感の量で反撃力変動
		IF NOWEX:絶頂数合計 && RAND:(1 + NOWEX:絶頂数合計)
			PRINTL 全心地忍耐著、根本提不起叛逆氣力……
			LOCAL = 10
		ELSEIF SOURCE:疼痛 > MAX(PALAM_LV(ABL:受虐属性), 500) && RAND:3
			PRINT 所帶來的
			SIF SOURCE:疼痛 >= 3000
				PRINT 過於激烈的
			PRINTL 劇痛勉強的忍耐著、完全生不起叛逆想法……
			LOCAL = 10
		ELSEIF MAX(CUP:快Ｃ, CUP:快Ｖ, CUP:快Ａ, CUP:快Ｂ, CUP:快Ｍ) >= 4000 && RAND:3
			PRINT 竭力忍耐、不由得發出苦悶的低呼的同時、
			LOCAL = 30
		ELSEIF (MAX(CUP:快Ｃ, CUP:快Ｖ, CUP:快Ａ, CUP:快Ｂ, CUP:快Ｍ) >= 1000 && RAND:3) || SOURCE:疼痛 >= 500
			PRINT 拼盡全力的忍耐著的同時、
			LOCAL = 50
		ELSE
			PRINT 像是完全投身於眼前的美景一樣、顯得毫不在意、
			LOCAL = 70
		ENDIF
	ENDIF

	IF LOCAL == 10

	ELSE
		SELECTCASE TEQUIP:六九式
		CASE TNUM("手淫")
			PRINTFORM 將%CALLNAME:PLAYER%的陰莖
			IF ABL:技巧 >= 3 && RAND:3 == 0
				PRINTFORML 嫻熟的調整著節奏持續擼動著……
			ELSEIF ABL:施虐属性 && RAND:3 == 0
				PRINTFORML 像是為了給予痛苦而不是快樂一樣、過於激烈地擼動著……
			ELSEIF ABL:欲望 + ABL:奉仕精神 >= 3 && RAND:2 == 0
				PRINTFORML 如同耽溺在夢中一樣的持續擼動著……
			ELSE
				PRINTFORML 用雙手盡心地侍奉著……
			ENDIF
		CASE TNUM("口交"), TNUM("深喉")
			PRINTFORM 將%CALLNAME:PLAYER%的陰莖
			IF TEQUIP:深喉
				;技巧４かつ奉仕精神３
				IF ABL:技巧 >= 4 && ABL:奉仕精神 >= 3 && RAND:3 == 0
					PRINTFORML 咽喉含住、讓啾嚕啾嚕的淫亂水聲與壓抑著的呻吟隨著頭頸的動作不斷迴響著……
				;奉仕精神3以上か欲情4以上
				ELSEIF (ABL:奉仕精神 >= 3 || PALAM:欲情 >= PALAMLV:4) && RAND:2 == 0
					PRINTFORML 整個口腔包裹住、如同耽溺在夢中一樣的品味著……
				ELSEIF  RAND:2 == 0
					PRINTFORML 整個口腔包裹住、持續激烈地吮吸著……
				ELSE
					PRINTFORML 整個口腔包裹后再深深地吞入咽喉中、借助本能地痙攣起來的咽喉軟肉緊緊包裹住陰莖……
				ENDIF
			ELSEIF ABL:信頼 <= 1 || ABL:奉仕精神 <= 1
				;娇小で追加
				SIF TALENT:娇小
					PRINT 小小的
				PRINTFORML 舌尖旋轉著舔舐著整個莖身……
			ELSEIF (ABL:信頼 >= 4 || ABL:奉仕精神 >= 3 || ABL:受虐属性 >= 3) && ABL:技巧 >= 3 && EXP:口交経験 >= 100 && RAND:3 == 0
				;技巧４かつ奉仕精神３
				IF ABL:技巧 >= 4 && ABL:奉仕精神 >= 3 && RAND:3 == 0
					PRINTFORML 咽喉含住、讓啾嚕啾嚕的淫亂水聲與壓抑著的呻吟隨著頭頸的動作不斷迴響著……
				;奉仕精神3以上か欲情4以上
				ELSEIF (ABL:奉仕精神 >= 3 || PALAM:欲情 >= PALAMLV:4) && RAND:2 == 0
					PRINTFORML 整個口腔包裹住、如同耽溺在夢中一樣的品味著……
				ELSEIF  RAND:2 == 0
					PRINTFORML 整個口腔包裹住、持續激烈地吮吸著……
				ELSE
					PRINTFORML 整個口腔包裹后再深深地吞入咽喉中、借助本能地痙攣起來的咽喉軟肉緊緊包裹住陰莖……
				ENDIF
			ELSE
				;奉仕精神3以上か欲情4以上
				IF  (ABL:奉仕精神 >= 3  || PALAM:欲情 >= PALAMLV:4) && RAND:3 >= 1
					PRINTFORML 整個口腔包裹住、持續激烈地吮吸著……
				ELSEIF  RAND:2 == 0
					PRINTFORML 整個口腔包裹後再深深地吞入咽喉中、借助咽喉的肉襞不斷地擼動著陰莖的尖端……
				ELSE
					PRINTFORML 整個口腔包裹住、費力地操作著口中的舌頭將陰莖前段的每一處都仔細地舔舐著……
				ENDIF
			ENDIF
		CASE TNUM("乳交")
			PRINTFORM 將%CALLNAME:PLAYER%的陰莖
			IF ABL:技巧 >= 3 && RAND:3 == 0
				PRINTFORM 嫻熟的調整著節奏、用自己那
			ELSEIF ABL:施虐属性 && RAND:3 == 0
				PRINTFORM 熱情地緊緊夾住、用自己那
			ELSEIF ABL:欲望 + ABL:奉仕精神 >= 3 && RAND:2 == 0
				PRINTFORM 如同耽溺在夢中一樣、用自己那
			ENDIF
			IF TALENT:貧乳
				PRINT 貧乏單薄的胸部
			ELSEIF TALENT:巨乳
				PRINT 豊満的
			ENDIF
			PRINTL 不斷地上下摩擦著……
		CASE TNUM("内褲冲")
			PRINTFORM 將%CALLNAME:PLAYER%的陰莖
			IF RAND:3 == 0
				IF CHECK_CLO("湿的") && RAND:2 == 0
					PRINT 濡濕的
				ELSEIF CHECK_CLO("闷热的")
					PRINT 蒸騰著熱汽的
				ENDIF
			ENDIF
			PRINTFORM 將%CLO("内裤冲")%
			IF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
				PRINTFORML 咯哧咯哧的擼動起來……
			ELSEIF RAND:2 == 0 && (TALENT:害羞 == 0 || TALENT:淫乱 || TALENT:恋慕)
				IF PENIS(TARGET)
					PRINTFORML 擼動起來…
				ELSE
					PRINTFORML %TEXTR("/用与小穴接触的部分包裹着/用中心的部分包裹着/用沾染了体液的部分")%擼動起來…
				ENDIF
			ELSE
				PRINTFORML 包裹著擼動起來……
			ENDIF
		CASE TNUM("奉仕舐陰")
			PRINTFORML 將%CALLNAME:PLAYER%不斷溢出甜蜜愛液的陰部投入地品味著……
		CASE TNUM("飞机杯冲")
			PRINTFORM 將%CALLNAME:PLAYER%的陰莖用飛機杯
			IF ABL:施虐属性 >= 3 && RAND:3 == 0
				PRINTFORM 套住後、用彷彿是想要捏斷一樣的大力
			ELSEIF TALENT:好奇心 && RAND:3 == 0
				PRINTFORM 套住後、在饒有興致的觀察著神態的同時、小心地
			ELSEIF MARK:屈服刻印 == 0 && RAND:3 == 0
				PRINTFORM 套住後、帶著相當複雜的表情
			ELSEIF RAND:3 == 0
				PRINTFORM 激烈地
			ELSEIF RAND:2 == 0
				PRINTFORM 噗嗤噗嗤地
			ELSE
				PRINTFORM 慢慢地、慢慢地
			ENDIF
			PRINTFORML 持續擼動著……
		ENDSELECT
	ENDIF

	SELECTCOM = TEQUIP:六九式
	CALL GET_EXP_COM

	;射精ゲージの増加
	CALL SAMEN_CHECK, LOCAL
;TARGETの行動　口交or乳交or奉仕舐陰
CASE 13, 14, TNUM("手淫") TO TNUM("乳交"), TNUM("内褲冲"), TNUM("奉仕舐陰"), TNUM("飞机杯冲"), TNUM("深喉")
	SELECTCOM = TEQUIP:PLAYER:六九式
	CALL GET_EXP_COM
ENDSELECT
SELECTCOM = MEMO_SELECTCOM

@COM_ORDER_COM99, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 10
	LOCALS = %LOCALS% - 害羞(10)
ENDIF
;献身的
IF TALENT:献身
	LOCAL += 6
	LOCALS = %LOCALS% + 献身(6)
ENDIF
;接受快感
IF TALENT:接受快感
	LOCAL += 3
	LOCALS = %LOCALS% + 接受快感(3)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF

;媚薬
IF TEQUIP:媚薬
	LOCAL += 6
	LOCALS = %LOCALS% + 媚薬(6)
ENDIF

;相互お掃除PLAY
IF CHECK_COM("清洁口交", TNUM("口交")) + CHECK_COM("清洁舔阴", TNUM("奉仕舐陰")) && CHECK_COM("清洁口交", TNUM("舐陰莖")) + CHECK_COM("清洁舔阴", 1)
	IF CHECK_COM("清洁口交", TNUM("口交"))
		LOCAL:1 = COM_ORDER_STAIN("Ｐ", "Ｍ")
	ELSE
		LOCAL:1 = COM_ORDER_STAIN("Ｖ", "Ｍ")
	ENDIF
ELSEIF PENIS(PLAYER)
	LOCAL:1 = COM_ORDER_STAIN("Ｐ", "Ｍ")
ELSE
	LOCAL:1 = COM_ORDER_STAIN("Ｖ", "Ｍ")
ENDIF

;汚れ有
IF LOCAL:1
	LOCAL -= LOCAL:1
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({LOCAL:1})
ENDIF

;合計を表示(43以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 43

RETURN RESULT


code:
@COM9
;舐肛※消費を少し高めにする
CALL PRINT_TRAIN_NAME(SELECTCOM)

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;-------------------------------------------------
;射精格子
;-------------------------------------------------
SIF TEQUIP:推倒
	CALL SAMEN_CHECK

RETURN 1

@FLAG_COM9, ARG
;対象のＡ⇔調教者の口的污垢が移動
CALL EQ_STAIN, TARGET, "Ａ", PLAYER, "Ｍ"

@SOURCE_COM9, ARG
CALL VARSET_COMF

CALL SOURCE_DOWNBASE_HP, 9

;※露出のソースを元より高めにしてみる
SOURCE:0:露出 = 800
SOURCE:0:逸脱 = 500
SOURCE:0:Ａ液体追加 = 100

;ABL:Ａ感覚をみる
SOURCE:0:Ａ快楽 = CALCF("Ａ刺激", 1, 70)
;調教者の舌靈活をみる
SOURCE:0:Ａ快楽 = SOURCE_EXABL(SOURCE:0:Ａ快楽, "舌靈活", PLAYER)

;面部騎乗の時には快感を得るよりはむしろ攻撃重視
SIF TEQUIP:推倒
	TIMES SOURCE:0:Ａ快楽, 0.30

;EXP:Ａ経験に応じたＡ快感、苦痛、恐怖、逸脱(反感)補正
SELECTCASE EXP:Ａ経験
CASE IS >= 300
	TIMES SOURCE:0:Ａ快楽, 1.60
	TIMES SOURCE:0:疼痛, 0.50
	TIMES SOURCE:0:恐惧, 0.20
	TIMES SOURCE:0:逸脱, 0.20
CASE IS >= 150
	TIMES SOURCE:0:Ａ快楽, 1.40
	TIMES SOURCE:0:疼痛, 0.60
	TIMES SOURCE:0:恐惧, 0.40
	TIMES SOURCE:0:逸脱, 0.60
CASE IS >= 50
	TIMES SOURCE:0:Ａ快楽, 1.20
	TIMES SOURCE:0:疼痛, 0.80
	TIMES SOURCE:0:恐惧, 0.70
CASE IS >= 20
	TIMES SOURCE:0:逸脱, 1.20
CASE IS >= 5
	TIMES SOURCE:0:Ａ快楽, 0.50
	TIMES SOURCE:0:疼痛, 1.50
	TIMES SOURCE:0:恐惧, 1.50
	TIMES SOURCE:0:逸脱, 1.50
CASEELSE
	TIMES SOURCE:0:Ａ快楽, 0.10
	TIMES SOURCE:0:疼痛, 2.00
	TIMES SOURCE:0:恐惧, 2.00
	TIMES SOURCE:0:逸脱, 2.00
ENDSELECT

;PALAM:欲情をみる
SELECTCASE PALAM:欲情
CASE IS < PALAMLV:1
	TIMES SOURCE:0:屈従, 0.30
CASE IS < PALAMLV:2
	TIMES SOURCE:0:屈従, 0.50
CASE IS < PALAMLV:3
	TIMES SOURCE:0:屈従, 0.80
CASE IS < PALAMLV:4
CASE IS >= PALAMLV:4
	TIMES SOURCE:0:屈従, 1.20
ENDSELECT

SIF TALENT:Ａ性向
	CALL CALC, "好みのPLAY", SOURCE:0:Ａ快楽

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM9
;同時推倒
IF TEQUIP:Ｗ推倒
	PRINTFORM %CALLNAME:TARGET%
	;助手が前回と同じ行動(違う行動の時は、ここでなく、EQUIP_COM32で記述)
	SIF SELECTCOM:1 == PREVCOM:1 && SELECTCOM:1 >= 0
		PRINTFORM %TEXTS("Ｗ推倒中")%、
	IF SELECTCOM == PREVCOM
		PRINTFORM 在%CALLNAME:MASTER%的臉上
		IF TALENT:不知羞恥 || TALENT:淫乱
			PRINT 將臀瓣更加的打開、
		ELSE
			PRINT 將腰肢輕輕下沉、
		ENDIF
		PRINTFORMW 讓尻穴被持續地舔舐侍奉著……
	ELSE
		IF COND("脚の使用") == 0
			PRINTFORMW 坐在%CALLNAME:MASTER%的臉頰上、讓不斷張合的肛門與期待已久的唇舌相吻……
		ELSE
			PRINTFORMW 跨在%CALLNAME:MASTER%的臉頰上、讓不斷張合的肛門與期待已久的唇舌相吻……
		ENDIF
		;淫乱or欲情4LV以上なら追加
		IF TALENT:淫乱 || PALAM:欲情 >= PALAMLV:4
			SIF (SELECTCOM:1 == 65 || SELECTCOM:1 == 75) && RAND:2 == 0
				PRINTFORMW %CALLNAME:TARGET%與%CALLNAME:(TARGET:1)%交纏著舌頭的同時、貪求著胯下傳來的甜美快感……
		ENDIF
	ENDIF
;被逆推ているときは面部騎乗
ELSEIF TEQUIP:推倒 && (TEQUIP:面部騎乗 == SELECTCOM || PREVCOM == SELECTCOM) && COND("前Ｔ調教者") == PLAYER
	PRINTFORM %CALLNAME:TARGET%
	IF RAND:2 == 0
		PRINTFORM %TEXTS("肛門")%
	ELSEIF ABL:欲望 + ABL:Ａ感覚 >= 6
		PRINTFORM 自己用手指張開
	ENDIF
	PRINTFORMW 肛門、讓%CALLNAME:PLAYER%持續用靈巧的舌頭侍奉著……
	;肛門が露出していないで、胖次をずらしていないなら自動的にずらす
	SIF !CHECK_CLO("Ａ露出")
		CALL SETFLAG, "ボトムを露出：テ接吻ト没有", TARGET
ELSEIF TEQUIP:推倒
	PRINTFORM %CALLNAME:TARGET%
	;前回が面部騎乗
	IF (TEQUIP:面部騎乗 == 1 || PREVCOM == 1) && COND("前Ｔ調教者") == PLAYER
		PRINTFORM 輕輕地半抬起腰肢、現在
		;肛門が露出していないで、胖次をずらしていないなら自動的にずらす
		SIF !CHECK_CLO("Ａ露出")
			CALL SETFLAG, "ボトムを露出：テ接吻ト没有", TARGET
	;足がある
	ELSEIF COND("脚が無い") == 0
		IF TALENT:害羞 && TALENT:内気 == 0
			PRINT 害羞地
		ELSEIF TALENT:内気 && TALENT:恋慕 == 0
			PRINT 胆怯、怯懦
		;「小悪魔」で随机1/2
		ELSEIF TALENT:小悪魔 && RAND:2 == 0
			PRINT 在臉上露出戲謔的壞笑的同時、
		ENDIF
		PRINTFORM 在%CALLNAME:PLAYER%的臉頰上方、
		IF (TALENT:害羞 || TALENT:内気) && RAND:3 == 0
			PRINTFORM 在用手隱約的遮擋著的同時
		;「傲慢」or「淫乱」or「小悪魔」で随机
		ELSEIF (TALENT:傲慢 || TALENT:淫乱 || TALENT:小悪魔) && RAND:2 == 0
			PRINT 大大地張開腿將股間風光暴露出來的同時
		ENDIF
		PRINT 快速坐下、
		;肛門が露出していないで、胖次をずらしていないなら自動的にずらす
		SIF !CHECK_CLO("Ａ露出") && CHECK_CLO("Ａ露出可")
			CALL SETFLAG, "ボトムを露出：テ接吻ト有些", TARGET
	;拉米亚、で後は一応他の条件もつける…拉米亚って胖次沒穿な
	ELSEIF COND("種族：拉米亚")
		IF TALENT:害羞 && TALENT:内気 == 0
			PRINT 害羞地
		ELSEIF TALENT:内気 && TALENT:恋慕 == 0
			PRINT 胆怯、怯懦
		;「小悪魔」で随机1/4
		ELSEIF TALENT:小悪魔 && RAND:4 == 0
			PRINT 在臉上露出戲謔的壞笑的同時、
		;「傲慢」or「淫乱」or「小悪魔」で随机1/3
		ELSEIF (TALENT:傲慢 || TALENT:淫乱 || TALENT:小悪魔) && RAND:3 == 0
			PRINT 為了更清楚的展示而慢慢地
		;欲情4以上＋A感覚3以上で随机1/2
		ELSEIF PALAM:欲情 >= PALAMLV:4 && ABL:Ａ感覚 >= 3 && RAND:2 == 0
			PRINT 興奮而迫不及待的
		;其他
		ELSE
			PRINT 順利的
		ENDIF
		PRINTFORM 讓仰視的%CALLNAME:PLAYER%的臉頰與腰肢快速接近、
	ENDIF
	;欲情6以上orA感覚3以上で随机1/2
	IF PALAM:欲情 >= PALAMLV:6 && ABL:Ａ感覚 >= 3 && RAND:2 == 0
		PRINTFORMW 讓%CALLNAME:PLAYER%的唇舌與%TEXTS("肛門")%肛門緊密相連……
	;それ以外
	ELSE
		PRINTFORMW 讓%CALLNAME:PLAYER%用舌頭輕輕刺入%TEXTS("肛門")%肛門……
	ENDIF
ELSEIF PREVCOM == SELECTCOM && COND("前Ｔ調教者") == PLAYER
	PRINTFORM %CALLNAME:PLAYER%將%CALLNAME:TARGET%的
	IF RAND:2 || TEQUIP:六九式
		PRINTFORM 將%TEXTS("お尻：感触")%%TEXTR("用臉頰填滿/用自己的臉頰享受/盡情享受")%之後、將
	ELSEIF COND("指の使用", PLAYER)
		PRINTFORM 將豐腴的尻肉撥開、將%TEXTR("變得一目了然的/被剝開暴露出來的/被推開來拉長的")%
	ENDIF
	PRINTFORMW %TEXTR("肛門執拗的舔舐著/尻穴那粘膜的異味用舌尖仔細品嚐著/肛門的皺褶用舌頭描繪著/尻穴的奇異韻味盡情享受著")%、如此沉醉了好一會……
;被逆推ていないときは从后面将…いや前からってのもあるか…
ELSE
	IF TEQUIP:六九式
		PRINTFORM %CALLNAME:PLAYER%將%CALLNAME:TARGET%的
		;肛門が露出していないで、胖次をずらしていないなら自動的にずらす
		IF !CHECK_CLO("Ａ露出") && CHECK_CLO("Ａ露出可")
			CALL SETFLAG, "ボトムを露出：テ接吻ト有些", TARGET
		ELSE
			PRINTFORM 豐腴的尻肉大力的撥開
		ENDIF
		PRINTFORM 、%TEXTS("６９")%
	ELSE
		SIF PLAYER != MASTER
			PRINTFORM %CALLNAME:PLAYER%将
		;前回Ａくぱぁ
		IF PREVCOM == 14
			PRINT 清楚地展示出肛門的
		;縄着用時かつ拉米亚でない
		ELSEIF TEQUIP:拘束 && COND("脚が無い") == 0
			PRINT 雙腿被拘束具強迫拉得大開的
		;害羞で快Ａor欲情が高い
		ELSEIF TALENT:害羞 && (PALAM:快Ａ >= PALAMLV:3 || PALAM:欲情 >= PALAMLV:4)
			PRINT 用雙手將因羞澀而染上紅暈的臉頰蒙上、膽怯地微微張開雙腿的
		;害羞で快Ａor欲情が低い
		ELSEIF TALENT:害羞 && (PALAM:快Ａ <= PALAMLV:3 || PALAM:欲情 <= PALAMLV:4)
			PRINT 因羞澀而抗拒的緊閉著雙腿的
		;淫乱or欲情4LV以上
		ELSEIF (TALENT:淫乱 || PALAM:欲情 >= PALAMLV:4) && ABL:Ａ感覚 >= 3
			PRINT 因渴求快感而誇張地張開雙腿的
		ENDIF
		PRINTFORM %CALLNAME:TARGET%的
		;肛門が露出していないで、胖次をずらしていないなら自動的にずらす
		IF !CHECK_CLO("Ａ露出") && CHECK_CLO("Ａ露出可")
			CALL SETFLAG, "ボトムを露出：テ接吻ト有些", TARGET
		;前から
		ELSEIF RAND:2
			PRINT 腰肢用力舉起
		;从后面将
		ELSE
			PRINT 股間用臉頰緊緊貼住
		ENDIF
		PRINT 、
	ENDIF
	PRINTFORM 將%TEXTS("肛門")%肛門用
	IF RAND:2 == 0
		PRINT 舌尖
	ELSE
		PRINT 嘴唇
	ENDIF
	IF RAND:4 == 0
		PRINT 温柔地
	ELSEIF RAND:3 == 0
		PRINT 謹慎認真地
	ELSEIF RAND:2 == 0
		PRINT 激烈地
	ELSE
		PRINT 像是在享受痛苦的反應一樣粗暴地
	ENDIF
	IF RAND:4 == 0
		PRINTFORMW 吮吸著……
	ELSEIF RAND:3 == 0
		PRINTFORMW 舔舐著……
	ELSEIF RAND:2 == 0
		PRINTFORMW 愛撫著……
	ELSE
		PRINTFORMW 品味著……
	ENDIF
ENDIF


code:
;実際に指令選択をしたら、これを呼ぶ
@SHOW_COM_ORDER_COM, ARG, ARG:1
;ARG:1が1ならテ接吻トが出力されない
TRYCALLFORM COM_ORDER_COM{ARG}, ARG:1
TCVAR:干劲 = RESULT

@PRINT_COM_ORDER_COM, ARGS, ARG, ARG:1, ARG:2
SELECTCASE ARG
CASE 0
	PRINTFORM %ARGS% = {ARG:1}
	SELECTCASE ARG:1
	CASE IS < ARG:2
		PRINT <
	CASE ARG:2
		PRINT =
	CASEELSE
		PRINT >
	ENDSELECT
	PRINTFORMW  実行値{ARG:2}
ENDSELECT

;RESULT > 0で実行値阴蒂ア
RETURN MAX(ARG:1 - ARG:2 + 1, 0)

@COM_ORDER, ARG
;少しは早くなるか…？
SIF TFLAG:実行値BOUNS
	RETURN TFLAG:実行値BOUNS
SIF TCVAR:基本実行値 && ARG
	RETURN TCVAR:基本実行値

STR:実行値 =
TCVAR:基本実行値 = 0

;周回BOUNS
IF FLAG:周回数
	IF FLAG:陥落人数
		SELECTCASE FLAG:陥落人数
		CASE IS <= 4
			TCVAR:基本実行値 += 1
			STR:実行値 = %STR:実行値% + 陥落人数BOUNS(1)
		CASE IS <= 8
			TCVAR:基本実行値 += 2
			STR:実行値 = %STR:実行値% + 陥落人数BOUNS(2)
		CASE IS <= 16
			TCVAR:基本実行値 += 3
			STR:実行値 = %STR:実行値% + 陥落人数BOUNS(3)
		CASE IS <= 24
			TCVAR:基本実行値 += 4
			STR:実行値 = %STR:実行値% + 陥落人数BOUNS(4)
		CASE IS <= 32
			TCVAR:基本実行値 += 5
			STR:実行値 = %STR:実行値% + 陥落人数BOUNS(5)
		CASE IS <= 40
			TCVAR:基本実行値 += 6
			STR:実行値 = %STR:実行値% + 陥落人数BOUNS(6)
		CASEELSE
			TCVAR:基本実行値 += 7
			STR:実行値 = %STR:実行値% + 陥落人数BOUNS(7)
		ENDSELECT
	ENDIF
	TCVAR:基本実行値 += FLAG:周回数
	STR:実行値 = %STR:実行値% + 周回数({FLAG:周回数})
ENDIF
IF EQUIP:Ｈ的奖励
	TCVAR:基本実行値 += 15
	STR:実行値 = %STR:実行値% + Ｈ的奖励(15)
ENDIF
IF TEQUIP:MASTER:无用相談
	TCVAR:基本実行値 += 8
	STR:実行値 = %STR:実行値% + 无用相談(8)
ENDIF
IF TEQUIP:解放感 && PLACE("野外")
	TCVAR:基本実行値 += 15
	STR:実行値 = %STR:実行値% + 解放感(15)
ENDIF
IF TEQUIP:欺诈师
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 欺诈师(5)
ENDIF
IF CFLAG:夸奖
	TCVAR:基本実行値 += 3
	STR:実行値 = %STR:実行値% + 心情好(3)
ENDIF
IF CFLAG:躁动难忍
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 躁动难忍中(5)
ENDIF
IF ABL:信頼
	TCVAR:基本実行値 += MIN(ABL:信頼, 10)
	STR:実行値 = %STR:実行値% + 信頼LV{ABL:信頼}({MIN(ABL:信頼, 10)})
ENDIF
IF ABL:欲望
	TCVAR:基本実行値 += MIN(ABL:欲望, 10)
	STR:実行値 = %STR:実行値% + 欲望LV{ABL:欲望}({MIN(ABL:欲望, 10)})
ENDIF

;刻印格子
IF MARK:苦痛刻印
	TCVAR:基本実行値 += MARK:苦痛刻印 * 5
	STR:実行値 = %STR:実行値% + 苦痛刻印LV{MARK:苦痛刻印}({MARK:苦痛刻印 * 5})
ENDIF
IF MARK:快楽刻印
	TCVAR:基本実行値 += MARK:快楽刻印 * 2
	STR:実行値 = %STR:実行値% + 快楽刻印LV{MARK:快楽刻印}({MARK:快楽刻印 * 2})
ENDIF
IF MARK:淫紋 && TALENT:男性 == 0
	TCVAR:基本実行値 += MARK:淫紋 * 1
	STR:実行値 = %STR:実行値% + 淫紋LV{MARK:淫紋}({MARK:淫紋 * 1})
ENDIF

IF MARK:屈服刻印
	TCVAR:基本実行値 += MARK:屈服刻印 * (3 + TALENT:高慢*3 - TALENT:謙虚)
	STR:実行値 = %STR:実行値% + 屈服刻印LV{MARK:屈服刻印}({MARK:屈服刻印 * (3 + TALENT:高慢*3 - TALENT:謙虚)})
ENDIF

;好感度(補正のMAXは16)
IF PLAYER == MASTER && MIN(GET_EXP_LV(CFLAG:好感度)/5, 16)
	TCVAR:基本実行値 += MIN(GET_EXP_LV(CFLAG:好感度)/5, 16)
	STR:実行値 = %STR:実行値% + 好感度({MIN(GET_EXP_LV(CFLAG:好感度)/5, 16)})
ENDIF

IF GET_PALAMLV(PALAM:欲情)
	TCVAR:基本実行値 += GET_PALAMLV(PALAM:欲情)
	STR:実行値 = %STR:実行値% + 欲情LV{GET_PALAMLV(PALAM:欲情)}({GET_PALAMLV(PALAM:欲情)})
ENDIF
IF GET_PALAMLV(PALAM:恐怖)
	TCVAR:基本実行値 += GET_PALAMLV(PALAM:恐怖)
	STR:実行値 = %STR:実行値% + 恐怖LV{GET_PALAMLV(PALAM:恐怖)}({GET_PALAMLV(PALAM:恐怖)})
ENDIF

IF TALENT:叛逆
	TCVAR:基本実行値 -= 10
	STR:実行値 = %STR:実行値% - 叛逆(10)
ENDIF
IF TALENT:堅強
	TCVAR:基本実行値 -= 5
	STR:実行値 = %STR:実行値% - 堅強(5)
ENDIF
IF TALENT:坦率
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 坦率(5)
ENDIF
IF TALENT:高慢 && TEQUIP:推倒 == 0
	IF COND("陶酔")
		TCVAR:基本実行値 -= 5
		STR:実行値 = %STR:実行値% - 高慢(陶酔)(5)
	ELSEIF COND("弛緩")
		TCVAR:基本実行値 -= 10
		STR:実行値 = %STR:実行値% - 高慢(弛緩)(10)
	ELSE
		TCVAR:基本実行値 -= 15
		STR:実行値 = %STR:実行値% - 高慢(15)
	ENDIF
ENDIF
IF TALENT:謙虚
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 謙虚(5)
ENDIF
IF TALENT:喜歡引人注目
	TCVAR:基本実行値 += 2
	STR:実行値 = %STR:実行値% + 喜歡引人注目(2)
ENDIF
IF TALENT:好色
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 好色(5)
ENDIF
IF TALENT:淫乱
	TCVAR:基本実行値 += 10
	STR:実行値 = %STR:実行値% + 淫乱(10)
ENDIF

IF COND("性別嫌悪")
	TCVAR:基本実行値 -= 10 * COND("性別嫌悪")
	STR:実行値 = %STR:実行値% - 性別嫌悪({10 * COND("性別嫌悪")})
ENDIF

IF TALENT:相思相愛 && PLAYER == MASTER
	TCVAR:基本実行値 += 10
	STR:実行値 = %STR:実行値% + 相思相愛(10)
ELSEIF TALENT:恋慕 && PLAYER == MASTER
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 恋慕(5)
ENDIF

;酔酒
IF STATE("烂醉", TARGET)
	TCVAR:基本実行値 += 10
	STR:実行値 = %STR:実行値% + 烂醉(10)
ELSEIF STATE("酔酒", TARGET)
	TCVAR:基本実行値 += 3
	STR:実行値 = %STR:実行値% + 酔酒(3)
ENDIF

IF COND("種族：人形")
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 人形(5)
ENDIF
IF COND("疲労")
	TCVAR:基本実行値 -= 30
	STR:実行値 = %STR:実行値% - 疲労困憊(30)
ENDIF

;調教者の素質
IF TALENT:PLAYER:魅力
	TCVAR:基本実行値 += 3
	STR:実行値 = %STR:実行値% + 魅力(3)
ENDIF
IF TALENT:PLAYER:謎之魅力
	TCVAR:基本実行値 += TALENT:PLAYER:謎之魅力
	STR:実行値 = %STR:実行値% + %TALENT_NAME(GETNUM(TALENT, "謎之魅力"), TALENT:PLAYER:謎之魅力)%({TALENT:PLAYER:謎之魅力})
ENDIF
IF TALENT:PLAYER:鼓舞
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 鼓舞(5)
ENDIF
IF TEQUIP:室内露出PLAY
	TCVAR:基本実行値 -= 5
	STR:実行値 = %STR:実行値% - 室内露出(5)
ENDIF
IF TEQUIP:攝像機
	TCVAR:基本実行値 -= 10
	STR:実行値 = %STR:実行値% - 撮影中(10)
ENDIF
;IF TEQUIP:野外PLAY
;	IF COND("人里")
;		TCVAR:基本実行値 -= 10
;		STR:実行値 = %STR:実行値% - 野外(人里)(10)
;	ELSE
;		TCVAR:基本実行値 -= 5
;		STR:実行値 = %STR:実行値% - 野外(5)
;	ENDIF
;ENDIF
;推倒時のBOUNSを追加
IF TEQUIP:推倒
	TCVAR:基本実行値 += 10
	STR:実行値 = %STR:実行値% + 被逆推特別BOUNS(10)
ENDIF
IF TEQUIP:甜言蜜语
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 甜言蜜语(5)
ENDIF

IF COND("調教中")
	;情緒
	IF MOOD() / 15
		TCVAR:基本実行値 += MOOD() / 15
		STR:実行値 = %STR:実行値% + 情緒({MOOD() / 15})
	ENDIF
	;指导者関係
	IF ASSI && TARGET != ASSI && STATE("撃沈", ASSI) == 0
		IF TEQUIP:動画撮影
			TCVAR:基本実行値 -= 5
			STR:実行値 = %STR:実行値% - スマホで撮影中(5)
		ENDIF
		IF TALENT:ASSI:鼓舞
			TCVAR:基本実行値 += 10
			STR:実行値 = %STR:実行値% + %TEXTS("助手の名称")%の鼓舞(10)
		ENDIF
		;相性格子
		SELECTCASE RELATION:(NO:ASSI)
		CASE 1 TO 30
			TCVAR:基本実行値 -= 30
			STR:実行値 = %STR:実行値% - %TEXTS("助手の名称")%との相性が最悪(30)
		CASE 31 TO 50
			TCVAR:基本実行値 -= 15
			STR:実行値 = %STR:実行値% - %TEXTS("助手の名称")%との相性が悪い(15)
		CASE 51 TO 80
			TCVAR:基本実行値 -= 5
			STR:実行値 = %STR:実行値% - %TEXTS("助手の名称")%との相性がいまいち(5)
		CASE 120 TO 149
			TCVAR:基本実行値 += 1
			STR:実行値 = %STR:実行値% + %TEXTS("助手の名称")%との相性なかなか(1)
		CASE 150 TO 199
			TCVAR:基本実行値 += 5
			STR:実行値 = %STR:実行値% + %TEXTS("助手の名称")%との相性が良好(5)
		CASE IS >= 200
			TCVAR:基本実行値 += 10
			STR:実行値 = %STR:実行値% + %TEXTS("助手の名称")%との相性が最高(10)
		ENDSELECT
	ENDIF
ELSE
	IF CFLAG:甜言蜜语
		TCVAR:基本実行値 += 5
		STR:実行値 = %STR:実行値% + 心动不已(5)
	ENDIF
	IF CHARA_NUM("室内の邪魔者")
		TCVAR:基本実行値 -= 20
		STR:実行値 = %STR:実行値% - 二人きりじゃない(20)
	ELSEIF PLACE("自宅") == 0 && CHARA_NUM("集合予定") > 2
		TCVAR:基本実行値 -= 10
		STR:実行値 = %STR:実行値% - まだまだ人が来る予定がある(10)
	ENDIF

	;換装のお願い中
	IF COND("胖次交易中")
		IF CHECK_CLO("胖次を脱いだらＶ露出")
			TCVAR:基本実行値 -= 20
			STR:実行値 = %STR:実行値% - 胖次を脱いだらＶ露出(20)
		ELSEIF CHECK_CLO("胖次を脱掉のが難儀")
			TCVAR:基本実行値 -= 5
			STR:実行値 = %STR:実行値% - 胖次を脱ぎづらい(5)
		ENDIF
	ELSEIF COND("内衣姿取引中")
		;上下揃った内衣姿
		IF EQUIP:胖次 && (EQUIP:胸罩 || TALENT:男性)
			TCVAR:基本実行値 -= 10
			STR:実行値 = %STR:実行値% - 内衣姿になって欲しい(10)
		;没穿外衣
		ELSEIF EQUIP:胖次 && EQUIP:胸罩 == 0
			TCVAR:基本実行値 -= 15
			STR:実行値 = %STR:実行値% - 脱いだら没穿外衣(15)
		;没穿下衣
		ELSEIF EQUIP:胖次 == 0 && EQUIP:胸罩
			TCVAR:基本実行値 -= 20
			STR:実行値 = %STR:実行値% - 脱いだらＶ露出(20)
		;全裸じゃん！
		ELSE
			TCVAR:基本実行値 -= 30
			STR:実行値 = %STR:実行値% - 脱いだら全裸(30)
		ENDIF
	ELSEIF COND("全裸交易中")
		TCVAR:基本実行値 -= 30
		STR:実行値 = %STR:実行値% - 裸体希望(30)
	ENDIF
	IF COND("胖次交易中") || COND("内衣姿取引中") || COND("全裸交易中")
		IF CADV("原生主义者", TARGET)
			TCVAR:基本実行値 += 15
			STR:実行値 = %STR:実行値% + 原生主义者(15)
		ELSEIF CADV("原生主义者")
			TCVAR:基本実行値 += 15
			STR:実行値 = %STR:実行値% + %CALLNAME:ASSI%の助言(15)
		ENDIF
	ENDIF
	IF COND("合意：露出PLAY") && COND("胖次交易中") + COND("内衣姿取引中") + COND("全裸交易中")
		TCVAR:基本実行値 += 10
		STR:実行値 = %STR:実行値% + 露出PLAYの合意(10)
	ENDIF
ENDIF

;現在地による補正
IF PLACE("野外")
	TCVAR:基本実行値 -= 20
	STR:実行値 = %STR:実行値% - 野外(20)
ELSEIF PLACE("partner的房間")
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 自分的房間(5)
ELSEIF PLACE("自室")
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + %CALLNAME:MASTER%的房間(5)
ELSEIF PLACE("情人旅館")
	TCVAR:基本実行値 += 5
	STR:実行値 = %STR:実行値% + 情人旅館(5)
ELSEIF PLACE("個室") == 0
	TCVAR:基本実行値 -= 5
	STR:実行値 = %STR:実行値% - 不在個室(5)
ENDIF

SIF ARG == 0
	PRINTFORM %SUBSTRING(STR:実行値, 3)%

RETURN TCVAR:基本実行値

;ARGSにPLAYERの場所、ARGS:1にTARGETの場所
@COM_ORDER_STAIN(ARGS, ARGS:1)
#FUNCTION
#DIM STAIN_TARGET
#DIM STAIN_PLAYER

;汚臭無視
SIF TALENT:汚臭無視
	RETURNF 0

SELECTCASE ARGS
CASE "Ｍ", "手", "脚", "Ｐ", "Ｖ", "Ａ", "Ｂ"
	STAIN_PLAYER = STAIN:PLAYER:ARGS
CASE "尿道"
	IF PENIS(PLAYER)
		STAIN_PLAYER = STAIN:PLAYER:Ｐ
	ELSE
		STAIN_PLAYER = STAIN:PLAYER:Ｖ
	ENDIF
CASEELSE
	RETURNF 0
ENDSELECT

SELECTCASE ARGS:1
CASE "Ｍ", "手", "脚", "Ｐ", "Ｖ", "Ａ", "Ｂ"
	STAIN_TARGET = STAIN:(ARGS:1)
CASE "尿道"
	IF PENIS(TARGET)
		STAIN_TARGET = STAIN:Ｐ
	ELSE
		STAIN_TARGET = STAIN:Ｖ
	ENDIF
CASEELSE
	STAIN_TARGET = 0
ENDSELECT

VARSET LOCAL

;愛液的污垢
IF CHECK_STAIN(STAIN_PLAYER, "Ｖ")
	LOCAL:0 = 16
	SIF COND("性別嫌悪")
		LOCAL:0 += 5 * 2
	SIF TALENT:男性 || TALENT:兩面通吃
		LOCAL:0 -= 8
	SIF TALENT:气味弱点
		LOCAL:0 -= 2
	;場所自体がＶで出浴ならペナ軽減
	SIF LOCAL:0 > 0 && ARGS == "Ｖ" && TEQUIP:PLAYER:出浴
		LOCAL:0 /= 2
ENDIF
;阴茎的污垢
IF CHECK_STAIN(STAIN_PLAYER, "Ｐ")
	LOCAL:1 = 5
	SIF COND("性別嫌悪")
		LOCAL:1 += 5 * COND("性別嫌悪")
	SIF TALENT:气味弱点
		LOCAL:1 -= 2
	;場所自体がＰで出浴ならペナ軽減
	SIF LOCAL:1 > 0 && ARGS == "Ｐ" && TEQUIP:PLAYER:出浴
		LOCAL:1 /= 2
	SIF CALCF("阴茎好き度", TARGET) > 0
		LOCAL:1 -= CALCF("阴茎好き度", TARGET)
ENDIF
;精液的污垢(精液中毒で相殺)
IF CHECK_STAIN(STAIN_PLAYER, "精液")
	SIF COND("性別嫌悪")
		LOCAL:2 += 5 * COND("性別嫌悪")
	SIF ABL:精液中毒
		LOCAL:2 = 6 - MIN(ABL:精液中毒, 5)*2 - (TALENT:气味弱点 > 0)
	SIF TALENT:吸精体質
		LOCAL:2 -= 3
	SIF TALENT:精液嗜好
		LOCAL:2 -= 6
ENDIF
;肛門的污垢
IF CHECK_STAIN(STAIN_PLAYER, "Ａ")
	LOCAL:3 = 65 - MIN(ABL:受虐属性, 5)*5 - MIN(ABL:施虐属性, 5)*2 - MIN(ABL:Ａ感覚, 10)
	SIF TALENT:尻穴狂い
		LOCAL:3 -= 20
	SIF TALENT:气味弱点 && TALENT:恋慕
		LOCAL:3 /= 2
	;場所自体がＡで出浴ならペナ軽減
	SIF LOCAL:3 > 0 && ARGS == "Ａ" && TEQUIP:PLAYER:出浴
		LOCAL:3 /= 2
ENDIF
;母乳的污垢
IF CHECK_STAIN(STAIN_PLAYER, "乳")
	LOCAL:4 += 1
ENDIF
;尿的污垢
IF CHECK_STAIN(STAIN_PLAYER, "尿")
	LOCAL:5 += 15
	SIF TALENT:气味弱点 && TALENT:恋慕
		LOCAL:5 /= 2
ENDIF
;血液汚れ
IF CHECK_STAIN(STAIN_PLAYER, "破瓜之血")
	LOCAL:6 += 30
	SIF COND("種族：吸血鬼")
		LOCAL:6 /= 2
ENDIF
;阴部の恥垢汚れ
IF CHECK_STAIN(STAIN_PLAYER, "Ｖ恥垢")
	LOCAL:7 = 20
	SIF COND("性別嫌悪")
		LOCAL:7 += 5 * COND("性別嫌悪")
	SIF TALENT:气味弱点 && TALENT:恋慕
		LOCAL:7 /= 2
	;場所自体がＶで出浴ならペナ軽減
	SIF LOCAL:7 > 0 && ARGS == "Ｖ" && TEQUIP:PLAYER:出浴
		LOCAL:7 /= 2
ENDIF
;ちんかす的污垢
IF CHECK_STAIN(STAIN_PLAYER, "Ｐ恥垢")
	LOCAL:8 = 20
	SIF COND("性別嫌悪")
		LOCAL:8 += 5 * COND("性別嫌悪")
	SIF TALENT:气味弱点 && TALENT:恋慕
		LOCAL:8 /= 2
	;場所自体がＰで出浴ならペナ軽減
	SIF LOCAL:8 > 0 && ARGS == "Ｐ" && TEQUIP:PLAYER:出浴
		LOCAL:8 /= 2
	IF CALCF("阴茎好き度", TARGET) > 0
		LOCAL:8 /= CALCF("阴茎好き度", TARGET)
		SIF TALENT:献身
			LOCAL:8 -= 5
		SIF TALENT:倒錯的
			LOCAL:8 -= 5
	ENDIF
ENDIF

;もともとある汚れは無視する＆場所によっては汚れの強さＵＰ
SELECTCASE ARGS:1
CASE "Ｍ"
	LOCAL:3 = MIN(MULTIPLY(LOCAL:3, 150), LOCAL:3 + 20)
CASE "Ｐ"
	LOCAL:0 = 0
	LOCAL:1 = 0
	LOCAL:2 = 0
	LOCAL:5 = 0
CASE "Ｖ"
	LOCAL:0 = 0
	LOCAL:1 = 0
	LOCAL:3 = MIN(MULTIPLY(LOCAL:3, 200), LOCAL:3 + 50)
CASE "Ａ"
	LOCAL:0 = 0
	LOCAL:1 = 0
	LOCAL:3 = 0
CASE "尿道"
	LOCAL:5 = 0
	LOCAL:3 = MIN(MULTIPLY(LOCAL:3, 200), LOCAL:3 + 50)
ENDSELECT

LOCAL = LOCAL:0 + LOCAL:1 + LOCAL:2 + LOCAL:3 + LOCAL:4 + LOCAL:5 + LOCAL:6 + LOCAL:7 + LOCAL:8

SIF TALENT:汚臭鈍感
	LOCAL = MAX(DIVIDE(LOCAL, 200), LOCAL - 20)
SIF TALENT:汚臭敏感
	LOCAL = MIN(MULTIPLY(LOCAL, 200), LOCAL + 20)

;陶酔中は汚れに鈍感になる
SIF LOCAL > 0 && COND("陶酔")
	LOCAL /= 2

LOCAL = MAX(LOCAL, 0)

RETURNF LOCAL